
Abstract

Thiswhitepapergives an introduction to formulatingand solvingRobustOptimizationproblemswith FICO R©XpressOp-

timization.

Robust optimization is a modelling framework that can be used to model optimization problems under uncertainty in

the input data. The uncertainty is assumed to be defined by a set of values that the uncertain data can take; this un-

certainty set can be described by linear constraints, convex quadratic constraints, or as a discrete set of vectors of

uncertain data. The solutions generated by a robust optimization model are feasible irrespective of the actual value

that the uncertain data will take.

The introductory part explains the general concepts of RobustOptimization andwhich types of formulations are avail-

able with Xpress. The remainder of this document is formed by a collection of example problems showing typical uses

of Robust Optimization in practice with a discussion of the problem implementation with Xpress Mosel.

The examples presented in this whitepaper are included in the examples/Mosel/robust directory of the Xpress

installation.

Introduction

1 Introduction
Robust optimization is a modelling paradigm that offers solutions when uncertainty in the input data can

be bounded within a well described region.

Robust optimization is different from stochastic optimization. While stochastic optimization usually aims

to identify a solution whose expected value of the objective function is optimal with respect to the

probability distribution of the uncertain data, robust optimization focuses on finding a solution that is

feasible regardless of the realization of the uncertain values; hence the term robust. When uncertainty

is associated with the objective of a model, robust optimization returns a solution that is optimal with

respect to the worst case of all realizations of the uncertain quantities.

1.1 Uncertains and robust constraints
Amodel quantity or coefficient whose value is subject to uncertainty is called an uncertain.

Intuitively, an uncertain can be viewed as a unknown quantity that is not under our control, but is

controlled by an opponent. The opponent makes his decision for the values of the uncertains after we

havemade our decision, i.e., after the solver has found an optimal value for themodel variables. Hence,

the values of the model variables must assume the worst case.

Amodel constraint that includes some uncertainty in the form of an uncertain coefficient or right hand

side is called a robust constraint. A good way to visualize the concept of robust optimization is to

consider a robust constraint as a two-phase expression. For the sake of argument, let us assume a

less-than or equal-to constraint that contains some uncertain model quantities:

a1x1 + ...+ alxl + (ak + uk)xk + ...+ (an + un)xn 6 b.

Here, all ai coefficients are known. However, the values of coefficients of variables xk to xn are only

known to some level of uncertainty, denoted by uk to un. As for traditional variables, wemust define the

values these uncertains can take.

The feasible region of the uncertains is called the uncertainty set, and is modelled by means of

constraints on the uncertains. In the example above, wemay know that the uncertain quantities are

likely to be small and their normmight be bounded from above, hence we want a solution that is robust

at a given confidence level. This implies a constraint on the uncertains of the form

u2
k + ...+ u2

n 6 r.

Having described the possible values of the uncertains, robust optimization aims to find a solution that is

always feasible, thus the robust constraint is equivalent to the constraint

a1x1 + ...+ alxl +maxu2
k+...+u2

n6r[(ak + uk)xk + ...+ (an + un)xn] 6 b.

A robust model may contain several robust constraints and several uncertainty sets. The solvability of a

robust optimization problem depends on whether the robust constraints in the model can be

transformed into a form that can be solved by the available mathematical programming solvers. The

resulting transformedmodel that is solved is called the robust counterpart.

1.2 Types of robust constraints
A robust constraint can only contain uncertains either multiplied by a variable or at the right-hand side.

The type of a robust constraint is defined by the type of uncertainty set (or sets) which define the

uncertain parameters used in the constraint. We shall work with a small knapsack problem example and

explore alternative formulations of the uncertainty set.

model Knapsack
uses "mmrobust" ! Load the robust library

parameters
NUM=5 ! Number of items
MAXVAL=100 ! Maximum value
MAXWEIGHT=80 ! Maximum weight
WTMAX=102 ! Max weight allowed for haul
end-parameters

declarations
Items=1..NUM ! Index range for items
VALUE: array(Items) of real ! Value of items
WEIGHT: array(Items) of real ! Weight of items
x: array(Items) of mpvar ! Decision variables
end-declarations

setrandseed(5);
forall(i in Items) do
VALUE(i):=50+random*MAXVAL
WEIGHT(i):=1+random*MAXWEIGHT
end-do
forall(i in Items) x(i) is_binary ! All x are 0/1

MaxVal:= sum(i in Items) VALUE(i)*x(i) ! Objective: maximize total value
WtMax:= sum(i in Items) WEIGHT(i)*x(i) <= WTMAX ! Weight restriction

maximize(MaxVal)

writeln("Solution:\n Objective: ", getobjval)
writeln("Item Weight Value")
forall(i in Items)

writeln(i, ": ", getsol(x(i)), strfmt(WEIGHT(i),8,2), strfmt(VALUE(i),8,2))

end-model

This basic model without uncertainty solves to 256.601. Its execution results in the following output

display.

Objective: 256.601
Item Weight Value
1: 0 74.37 118.04
2: 1 62.34 141.75
3: 1 27.74 114.85
4: 0 53.17 100.60
5: 0 74.02 65.82

1.2.1 Simple bounds on the uncertain coefficients

It is possible to impose simple box constraints on the uncertain values. This is often an immediate

approach as it simply defines the range of values a coefficient can take, on a per-coefficient basis.

Uncertains are introduced to the model as follows.

parameters
NUM=5 ! Number of items
MAXVAL=100 ! Maximum value
MAXWEIGHT=80 ! Maximum weight
WTMAX=102 ! Max weight allowed for haul
WTPERCENT=0.3 ! Uncertainty as a percentage
end-parameters

Robust Optimization with Xpress Fair Isaac Corporation Proprietary Information 4

declarations
Items=1..NUM ! Index range for items
VALUE: array(Items) of real ! Value of items
WEIGHT: array(Items) of real ! Weight of items
x: array(Items) of mpvar
WeightUncertainty: array(Items) of uncertain ! Uncertains representing

! deviation from weight
end-declarations

forall(i in Items) x(i) is_binary ! All x are 0/1

setrandseed(5);
forall(i in Items) do
VALUE(i):=50+random*MAXVAL
WEIGHT(i):=1+random*MAXWEIGHT
end-do

MaxVal:= sum(i in Items) VALUE(i)*x(i) ! Objective: maximize total value

forall(i in Items) do
WeightUncertainty(i) <= WTPERCENT*WEIGHT(i) ! Uncertainty is a percentage of

! the expected weight
WeightUncertainty(i) >= 0 ! and only expected to go up this time
end-do
WtMax:= sum(i in Items) (WEIGHT(i) + WeightUncertainty(i))*x(i) <= WTMAX

! Weight restriction

maximize(MaxVal)

Note the uncertain type for the array WeightUncertainty. This type is used in Mosel to define a

parameter that is not known a priori by the user and is not under the user’s control. The constraint

WtMax that previously was of the type linctr now becomes a robustctr (the declaration of
constraints is optional). The newmodel allows a 30%maximum deviation from the expected weight.

Solving it we get 141.751.

Objective: 141.751
Item Weight Value
1: 0 74.37 118.04
2: 1 62.34 141.75
3: 0 27.74 114.85
4: 0 53.17 100.60
5: 0 74.02 65.82

The solution has shifted to pick the most valuable item only. Increasing the weight of the selected item

by the allowed 30% it becomes 81.042. The optimal solution is robust, i.e., for any binary solution with a

larger objective function there exists a vector of uncertains that would violate the robust constraint. The

objective has decreased significantly because of uncertainty; the difference in objective function values

between the original problem and the robust version is often referred to as the price of robustness [?]. If

a subset of the uncertainty set is used, for example with a smaller value of WTPERCENT, the effect of
uncertainty decreases and we can obtain a (still robust) solution with a larger objective function. On the

contrary, increasing WTPERCENT is equivalent to increasing uncertainty, which reduces the objective
function value of the robust optimal solution.

Note the explicit non-negativity restrictions on the uncertain coefficients. There is no default lower

bound imposed on uncertains, in contrast to traditional decision variables; this is to reflect that

uncertainty is not typically biased toward positive values.

Even if a robust modelling feature makes solving such amodel simple, it is important to realize that

when each individual uncertain is independently bounded by simple bounds only, the samemodel can be

derived bymodifying all coefficients in the way that restricts the constraint the most. In the knapsack

example, as all variables must be nonnegative and all weights are also non-negative, this means adding

the upper bound of the uncertains to the original coefficients.

declarations
Items=1..NUM ! Index range for items
VALUE: array(Items) of real ! Value of items
WEIGHT: array(Items) of real ! Weight of items
x: array(Items) of mpvar ! Decision variables
end-declarations

forall(i in Items) x(i) is_binary ! All x are 0/1

setrandseed(5);
forall(i in Items) do
VALUE(i):=50+random*MAXVAL
WEIGHT(i):=1+random*MAXWEIGHT
end-do

MaxVal:= sum(i in Items) VALUE(i)*x(i) ! Objective: maximize total value
WtMax:= sum(i in Items) WEIGHT(i)*(1+WTPERCENT)*x(i) <= WTMAX

! Weight restriction

maximize(MaxVal)

Indeed, the solution remains 141.751.

Objective: 141.751
Item Weight Value
1: 0 74.37 118.04
2: 1 62.34 141.75
3: 0 27.74 114.85
4: 0 53.17 100.60
5: 0 74.02 65.82

Realization of uncertain parameters Recall that an uncertain is an unknown quantity that is not under

our control, but is rather chosen by a third party “opponent”. Solving a robust optimization problem

amounts to solving a leader-follower game in which the user is the leader and the opponent obtains the

variables from the user and chooses the uncertain parameters.

In some cases, after solving a robust optimization problem it is possible to obtain the value of the

uncertain parameters that correspond to the robust solution. For this, it is necessary to have an optimal

solution −x to the robust problem. Consider the robust constraint

∑
i=1

aiuixi 6 b

and assume that the uncertains ui are within the ellipsoidal uncertainty set defined by

∑
i=1

u2
i 6 4

The opponent implicitly chooses the value of the uncertain so that the left-hand side of the robust

constraint ismaximized (against the6 constraint). Hence a realization of the uncertains for this robust

constraint can be found bymaximizing
∑

i=1 a
−
i xiui subject to the quadratic uncertainty constraint∑

i=1 u
2
i 6 4. It is possible to obtain the values of the uncertain parameters through the same function

used to obtain the variable, getsol. The modalities for use of getsol are explained in the Mosel

manual. Note that getsol can be used on uncertains only after the robust counterpart has been
solved.

1.2.2 Linear constraints on the uncertainty sets

Bound constraints are only one example of the possible uncertainty sets we canmodel. Let us suppose

that some or all of the uncertains can be aggregated and that we can write a single uncertain constraint

with all of them. Let us refine the example from the simple bounds case, and require instead that the

sum of the uncertains is at most 30% of the total weight.

parameters
NUM=5 ! Number of items
MAXVAL=100 ! Maximum value
MAXWEIGHT=80 ! Maximum weight
WTMAX=102 ! Max weight allowed for haul
WTPERCENT=0.3 ! Uncertainty as a percentage
end-parameters

declarations
Items=1..NUM ! Index range for items
VALUE: array(Items) of real ! Value of items
WEIGHT: array(Items) of real ! Weight of items
x: array(Items) of mpvar ! Decision variables
WeightUncertainty: array(Items) of uncertain ! Uncertains representing

! deviation from weight
end-declarations

forall(i in Items) x(i) is_binary ! All x are 0/1

setrandseed(5);
forall(i in Items) do
VALUE(i):=50+random*MAXVAL
WEIGHT(i):=1+random*MAXWEIGHT
end-do

MaxVal:= sum(i in Items) VALUE(i)*x(i) ! Objective: maximize total value

forall(i in Items) do
WeightUncertainty(i) >= 0
end-do

sum(i in Items) WeightUncertainty(i) <= WTPERCENT * sum(i in Items) WEIGHT(i)
WtMax:= sum(i in Items) (WEIGHT(i) + WeightUncertainty(i))*x(i) <= WTMAX

! Weight restriction

maximize(MaxVal)

While this sounds reasonable, because of the large right-hand side (30% of the total weight) the

modifiedmodel is subject to a very conservative uncertainty set and will solve to a solution of all zeros.

Objective: 0
Item Weight Value
1: 0 74.37 118.04
2: 0 62.34 141.75
3: 0 27.74 114.85
4: 0 53.17 100.60
5: 0 74.02 65.82

This behavior highlights how robust optimization works. Whenmoving from the previous uncertainty

set, with a small upper bound on the uncertains, to the new uncertainty set, wemigh in fact relax the

uncertainty set: if WTPERCENT * sum (i in Items) WEIGHT (i) is large enough, we havemade

the feasible space of the opponent significantly larger. This makes it possible that all uncertainty can be

placed on a single item. Since each item can be too heavy even individually, the all zero solution becomes

the only feasible one. This is a very important note to the behavior of robust optimization. When the

uncertainty set is relaxed, the robust counterpart becomesmore restrictive. Vice versa, to relax a robust

constraint (andmake it less conservative), its corresponding uncertainty set needs to be tightened.

A solution to the model in the example is then to restrict the feasible space of the uncertain values, that

is, to tighten the uncertainty set by adding new uncertain constraints. In fact, these constraints would

need to limit the amount of uncertainty per coefficient andwill resemble the simple bound case in effect.

An uncertainty set defined by a set of linear constraints is often referred to as a polyhedral uncertainty

set. It is important to note that polyhedral uncertainty sets have very important applications, but care

must be taken when applying them as the example above has shown.

1.2.3 Ellipsoidal uncertainty sets

Aswe have seen with the previous example, the robust counterpart or opponent can take full advantage

of the vertex solutions of a polyhedral uncertainty set making the model more conservative than

possibly intended. This could be countered by either restricting such constraints or by refining the set,

for example adding new linear constraints. It is easy to see that this is an approximation only to the

underlying problem, andmay be impractical in larger dimensions.

If the uncertainty set is known to be described by a quadratic constraint, for instance we know that the

norm of the vector of the uncertains is bounded from above, then we can use ellipsoidal uncertainty

sets. Another use for ellipsoidal uncertainty comes when the uncertainty set can be described by a

confidence ellipsoid that is defined by amean vector a, a covariance matrixQ, and a level of confidence

α. In this case, the uncertainty set is of the form (u− a)TQ(u− a) 6 α. In the followingmodel, we set α
to WTPERCENT * sum(i in Items) WEIGHT(i).

parameters
NUM=5 ! Number of items
MAXVAL=100 ! Maximum value
MAXWEIGHT=80 ! Maximum weight
WTMAX=102 ! Max weight allowed for haul
WTPERCENT=0.3 ! Uncertainty as a percentage
end-parameters

declarations
Items=1..NUM ! Index range for items
VALUE: array(Items) of real ! Value of items
WEIGHT: array(Items) of real ! Weight of items
x: array(Items) of mpvar ! Decision variables
WeightUncertainty: array(Items) of uncertain ! Uncertains representing

! deviation from weight
end-declarations

forall(i in Items) x(i) is_binary ! All x are 0/1

setrandseed(5);
forall(i in Items) do
VALUE(i):=50+random*MAXVAL
WEIGHT(i):=1+random*MAXWEIGHT
end-do

MaxVal:= sum(i in Items) VALUE(i)*x(i) ! Objective: maximize total value

sum(i in Items) WeightUncertainty(i)^2 <= WTPERCENT * sum(i in Items) WEIGHT(i)
WtMax:= sum(i in Items) (WEIGHT(i) + WeightUncertainty(i))*x(i) <= WTMAX

! Weight restriction

maximize(MaxVal)

This model gives the desired solution as shown below. Note the value of the uncertains as obtained after

maximization.

Objective: 215.445
Item Weight Value Uncertain
1: 0 74.37 118.04 0.00
2: 0 62.34 141.75 0.00
3: 1 27.74 114.85 6.61
4: 1 53.17 100.60 6.61
5: 0 74.02 65.82 0.00

Note that polyhedral uncertainty creates a robust counterpart that is of the same class as the original

problem, as it amounts to adding a set of linear constraints. This is not the case with ellipsoidal

uncertainty, which creates a robust counterpart that contains one second-order conic constraint for

each robust constraint subject to ellipsoidal uncertainty, and hence may change the nature of the

problem: if the original problem is a Linear Programming (LP) problem, the robust counterpart is a

Second Order Conic Programming (SOCP) problem; if the original problem is a Mixed Integer Linear

Programming (MILP) problem, the robust counterpart is a MISOCP.While solvers for MISOCP are now

more powerful, a MISOCP problem is generally more difficult to solve than aMILP.

We conclude this section with a note about the use of getsol on uncertains: the last column of the

output table shows the value of the uncertain WeightUncertainty for this particular solution. It is
easy to notice that the nonzero uncertain is in accordance with the nonzero of the solution. This squarely

aligns with viewing robust optimization as a leader-follower game: although the robust counterpart is

solved just once, it emulates the dynamics of the opponent, who chose to increase the uncertain to make

the robust constraint as tight as possible, and this can only be done by increasingWeightUncertainty(3)

andWeightUncertainty(4).

1.2.4 Equality constraints and uncertain values with no uncertain con-

straints

At this point it is worth mentioning some pitfalls of robust optimization that one needs to be careful

about. First, equality robust constraints are very restrictive in robust optimization.

declarations
x, y, z : mpvar
e : uncertain
end-declarations

e <= 1
e >= 0

x + y + e*z = 10

maximize(z)

In the solution to this problem, zwill always be zero. The reason is simple: whenever z is equal to a
non-zero value, if the value of variables x and y are fixed, any change in the value of the uncertain ewill
make the equality constraint infeasible. In general, the effect of uncertainty in equality robust

constraints is to project the feasible space of the original problem to a smaller dimensional one.

Second, it is always beneficial to make sure that the uncertainty set is bounded from both below and

above. The small example below shows the same effect as the case of the equality constraint.

declarations
x, y, z : mpvar
e : uncertain
end-declarations

x + y + e*z <= 10

maximize(z)

Here zwill again solve to zero, as for any nonzero value of z a large enough ewouldmake the constraint

infeasible. Often, such missing bounds lead to an infeasible robust optimization problem.

1.2.5 Mixing uncertainty sets and types

Due to the theory of robust optimization, there are limitations on how uncertains can be used in

uncertainty sets and robust constraints.

The most prominent restriction prohibits, by default, the use of the same uncertain values in different

robust constraints. This restriction applies also to amodel where two uncertains are used independently

in two robust constraints if these two uncertains are contained in a constraint defining an uncertainty

set. This restriction is enforced by virtue of how the corresponding robust counterparts are created,

which allows the opponent to select its strategy, i.e. the value of an uncertain, on a per-constraint basis.

Therefore, if the same uncertain appears in two robust constraints its value is allowed to take different

values in each robust constraint. However, it is often convenient to make use of this behavior if the same

uncertain set description is used for multiple robust constraints.

Consider the following example.

declarations
x, y: mpvar
e, f : uncertain
end-declarations

x*e <= 1
y*f <= 1

e >= 0
f >= 0
e+f <= 1

maximize(x+y)

The example will not solve using default settings, as the uncertain constraint e+ f 6 1 connects the two

robust constraints to the same uncertainty set, thereby they overlap. Setting

ROBUST_UNCERTAIN_OVERLAP to true will allow for the problem to be solved.

declarations
x, y: mpvar
e, f : uncertain
end-declarations

RobC1:= x*e <= 1
RobC2:= y*f <= 1

e >= 0
f >= 0
e+f <= 1

setparam("XPRS_PROBNAME","h")

setparam("ROBUST_UNCERTAIN_OVERLAP", true)

maximize(x+y)

However, the solution returned is x=1=y, which can easily be seen as non-optimal if e+ f 6 1 is satisfied:

the solution is assuming that e=1 to satisfy the first robust constraint and that f=1 to satisfy the second
one. Allowing the overlap will result in the opponent being able to optimize its decisions independently

for the two robust constraints. This is the reason why this mode is not allowed by default; although often

this is exactly what the modeller wants, setting ROBUST_UNCERTAIN_OVERLAP to true allows for

reusing the same uncertainty set definition multiple times without the need to repeat the uncertain

constraints. However, care must be taken as the uncertain values do not take a defined value upon

solving the problem and it might be different for different robust constraints.

The retrieval of uncertains that overlap across different constraints adds a further twist: the two

constraints xe 6 1 and yf 6 1 admit different combinations of the uncertains (e, f) as can be seen from
the output.

x = 1 y = 1
constraint RobC1: x*e <= 1, uncertains (e,f) = (1,0)
constraint RobC2: y*f <= 1, uncertains (e,f) = (0,1)

The first constraint can only be made tighter by increasing uncertain e, while the second, whose only
term is multiplied by f, can bemade tight by setting the latter to its maximum value. For this reason,

overlapping uncertains can only be retrieved by calling getsol passing one of the robust constraints
where they appear. The value of that uncertain will then be equal to what the opponent would choose if

that were the only robust constraint.

1.2.6 Cardinality restrictions for uncertains

Cardinality restrictions allow for limiting the number of uncertain values that are different from zero. In

other words, a cardinality constraint limits the number of uncertains that are non-zero or, more

generally, not at their nominal value. Due to the way in which the robust counterpart is created,

uncertains that appear in a cardinality restriction should always have reasonable lower and upper

bounds.

In the next example we show how to use uncertains in multiple robust constraints (i.e. in uncertains that

overlap). The example is a small knapsack type problem, where only 1 of the 3 items is allowed to differ

in weight from zero.

declarations
x,y,z: mpvar
ex, ey, ez: uncertain

end-declarations

x is_binary ; y is_binary ; z is_binary

20*x + 10*y + 5*z >= 20

ex>=0 ; ex<=1
ey>=0 ; ey<=1
ez>=0 ; ez<=1

cardinality({ex,ey,ez},1) ! Allow only one of the uncertains to be nonzero

setparam("ROBUST_UNCERTAIN_OVERLAP",true) ! Allow overlaps

10*(x-x*ex) + 10*(y-y*ey) + 10*(z-z*ez) >= 10
10*(x-x*ex) + 10*(y-y*ey) + 10*(z-z*ez) <= 20

maximize(x+y+z)

The solution returned is x=y=1 and z=0. It is important to emphasize that, as for the other constraints,

the cardinality restriction is taken into consideration on a per robust constraint basis, and the model

does not solve unless the ROBUST_UNCERTAIN_OVERLAP parameter is set to true.

1.2.7 Using historical data: scenarios

Scenarios are an efficient way of building robust optimization problems using historical realizations of

the uncertains. Suppose we have a vector u of uncertains. We do not know an uncertainty set that

would fit our model, but for the uncertain vector we have a database of records containing the value of

each uncertain for each month of the past 20 years. Therefore we would like a model where one or

more constraints are satisfied for each record of the uncertain vector in our database. Although we

may not have a good approximation of the uncertainty set, robustness against historical data might be

sufficient.

Declaring a scenario is equivalent to manually adding all the corresponding constraints for all

realizations to the problem; this capability is made available to make sure that the scenario data are

efficiently handled by the solver and to avoid the creation of an overly large problem.

Scenarios can be driven by the historical data, as shown in the next extension of the knapsack problem

analyzed before where a number of historical measurements have been taken.

parameters
NUM=5 ! Number of items
MAXVAL=100 ! Maximum value
MAXWEIGHT=80 ! Maximum weight
WTMAX=102 ! Max weight allowed for haul
UNCERTAINTY_LEVEL=0.3 ! How much we are uncertain about the weight
HISTORIC_PERIODS=100 ! Number of scenarios
end-parameters

declarations
Items=1..NUM ! Index range for items
VALUE: array(Items) of real ! Value of items
WEIGHT: array(Items) of real ! Weight of items
UncertainWeight:array(Items) of uncertain
x: array(Items) of mpvar ! 1 if we take item i; 0 otherwise
historical_weights: array(range, set of uncertain) of real
end-declarations

forall(i in Items) x(i) is_binary ! All x are 0/1

setrandseed(5);
forall(i in Items) do
VALUE(i):=50+random*MAXVAL
WEIGHT(i):=1+random*MAXWEIGHT
end-do

MaxVal:= sum(i in Items) VALUE(i)*x(i) ! Objective: maximize total value
WtMax:= sum(i in Items) (WEIGHT(i)+UncertainWeight(i))*x(i) <= WTMAX

! Generate historical data, this would be data collected from actual realizations
forall(period in 1..HISTORIC_PERIODS, i in Items)

historical_weights(period, UncertainWeight(i)) :=
WEIGHT(i)*UNCERTAINTY_LEVEL*random

! Generate a solution that would be feasible for ALL historic realizations
scenario(historical_weights)

maximize(MaxVal)

It is beneficial to examine the deterministic equivalent to a scenario based robust model. Consider the

following simple scenario basedmodel.

declarations
x, y : mpvar
e, f : uncertain
HISTDATA: array(range, set of uncertain) of real
end-declarations

! Load historical data for e and f
HISTDATA(1, e) := 1
HISTDATA(1, f) := 2

HISTDATA(2, e) := 3
HISTDATA(2, f) := 1
HISTDATA(3, e) := 3
HISTDATA(3, f) := 2

e*x + f * y <= 1

! Generate a solution that would be feasible for ALL historic realizations
scenario(HISTDATA)

maximize(x+y)

Its deterministic equivalent is:

declarations
x, y : mpvar
end-declarations

Scenario1 := 1 * x + 2 * y <= 1
Scenario2 := 3 * x + 1 * y <= 1
Scenario3 := 3 * x + 2 * y <= 1

maximize(x+y)

1.2.8 Uncertainty in the objective function

When uncertain coefficients are introduced in the objective, the solution to the robust counterpart

reflects the most conservative objective. More specifically, in a minimization problemwhere the

objective function is f(x;u), where x is a vector of variables and u is a vector of uncertains, the optimal

solution is one that minimizes the function g(x) = maxuf(x;u). This is equivalent to converting the
objective to the robust constraint z > f(x;u), where a new auxiliary variable z is used to represent the
objective.

1.3 Nominal values: centered and uncentered uncertainty
The nominal value of an uncertain represents its default value, a real-valued number which is the

expected or non-robust version of the uncertain. In the examples so far, all uncertainty has been

treated so that the nominal value of each uncertain is zero, and in effect the uncertain has been treated

as an error term. The default value or center of an uncertain can be shifted by specifying a nominal

value that is different from zero. Using nominal values has the advantage that they allow one to work on

both the robust optimizationmodel and on a deterministic one obtained by fixing the uncertains (to their

nominal values) in the samemodel.

Before detailing the use cases and rules when working with nominal values of uncertains, let us give an

example on centered and uncentered uncertains.

I. Using zero centered uncertainty, where an uncertain is modelled as a zero centered error, i.e.,

uncertainty with zero nominal value:

(5+ uncertain)x 6 1

is expressed as:

declarations
x : mpvar
u : uncertain

end-declarations

(5+u)*x <= 1

II. Uncertainty as coefficient, where the uncertain is modelled as a coefficient, i.e. an uncertain

coefficient with 5 as its nominal value:

(5+ uncertain)x 6 1

can also be expressed as:

declarations
x : mpvar
u : uncertain

end-declarations

u:= 5 ! Setting the nominal value
u*x <= 1

As the example above shows, a nominal value is set using the assignment operator ’:=’. It is important to

distinguish this from the equal operator ’=’, which would fix the value of the uncertain to the given value

instead.

1.3.1 The price of robustness

Either when using zero-centered uncertains or when setting up nominal values, it is possible to solve the

model with the uncertainty removed, i.e. all uncertain coefficients being fixed to their nominal values.

This makes it possible to calculate the cost of having uncertainty in the model (often referred to as the

price of robustness), as well as to check whether the model is feasible without the added uncertainty.

The next example shows how to solve with all uncertains fixed to their nominal values.

declarations
x : mpvar
u : uncertain

end-declarations

0 <= u
u <= 3

(1+u)*x <= 1

maximize(XPRS_NOMINAL, x)
nominal_objective := getobjval
writeln("Objective at nominal values:", nominal_objective );

maximize(x)
robust_objective := getobjval
writeln("Robust objective:", robust_objective );
writeln("Price of robustness:", nominal_objective - robust_objective );
writeln("Worst value of uncertain:", u.sol)

The output of the model is:

Objective at nominal values: 1
Robust objective: 0.25
Price of robustness: 0.75

When no nominal value is defined for an uncertain, the default nominal value of zero is used.

1.3.2 Working with nominal values

The workings of nominal values are defined by two simple rules.

Rule 1: setting a nominal value is shifting the domain of the uncertain.

Setting a nominal value for an uncertain to u:=a replaces all later occurrences of uwith u+a.

The following simple model solves to x = 0.5:

declarations
x : mpvar
u : uncertain
end-declarations

0 <= u
u <= 1

(1+u)*x <= 1

maximize(x)

as the worst case is when u takes its upper bound of 1.

Changing the nominal value of u to say 2, u is replaced by u+2 and the model solves to 0.25 instead.

declarations
x : mpvar
u : uncertain
end-declarations

0 <= u
u <= 1

u := 2 ! Shifts the center of the uncertain

(1+u)*x <= 1

maximize(x)

The uncertain u still takes its upper bound of 1, but the constraint loaded now reads (1+ u+ 2)x 6 1 with

u taking on its upper bound of 1.

Rule2: a changed nominal value only affects robust and uncertainty set constraints that are defined

after it has been assigned.

This behavior is best understood when put into analogy with how real-valued parameters behave.

Looking at the example of rule 1, this behavior is already observed: changing the nominal value has only

affected the robust constraint but not the bounds declared before this point.

declarations
x : mpvar
r : real
end-declarations

r := 3

C1 := (1+r)*x <= 1

r := 5

maximize(x)

Themodel solves to x = 0.25, that is the actual value of r at the time of declaring the constraint C1

has been used (3), and redefining it later had no effect on the constraint. This behavior is reproduced

when working with the nominal values of uncertains.

declarations
y : mpvar
u : uncertain
end-declarations

u := 3

(1+u)*y <= 1

u := 5

maximize(XPRS_NOMINAL, y)

The nominal version of the model where the uncertain coefficients are used solves (notice the nominal

argument to maximize) to the same as when real valued coefficients are used: y = 0.25.

These rules offer significant flexibility on how to use the nominal values for uncertains, but care must be

taken and the effects of rule 2 should always be kept in mind.

1.3.3 Using nominal values to shift the uncertainty set

Nominal values can be used to shift an uncertain coefficient while keeping its effect around the nominal

values the same. Consider the following example:

declarations
x,y : mpvar
e,f : uncertain
end-declarations

e^2 + f^2 <= 2 ! An ellipsoidal uncertainty constraint

e*x + f*y <= 1

maximize(x + y)

The solution to the model is x=0.5, y=0.5. The values of uncertains e and f can take their value from a

ball of radius
√
2. When nominal values are added, the model becomes:

declarations
x,y : mpvar
e,f : uncertain
end-declarations

e^2 + f^2 <= 2

e := 1
f := 1

e*x + f*y <= 1

maximize(x + y)

The solution moves to x=0.25, y=0.25. To explore the reasons for this change, let us substitute the effect

of the nominal values directly into the problem:

declarations
x,y : mpvar
e,f : uncertain

end-declarations

e^2 + f^2 <= 2

(e+1)*x + (f+1)*y <= 1

maximize(x + y)

The range of uncertains has remained the same, but in the robust constraint the values around which

they are centered shift accordingly to the nominal values.

1.3.4 Using nominal valued uncertains as coefficients

Uncertains can be assigned a nominal value before the first time they are used, however they may

behave in a way that is not expected. Consider three versions of the samemodel:

Case 1 Case 2 Case 3

No nominal values are set Nominal values are set to 1 Nominal values are set to

different values

declarations

x,y : mpvar

e,f : uncertain

end-declarations

e + f6 2

e*x + f*y6 1

maximize(x + y)

declarations

x,y : mpvar

e,f : uncertain

end-declarations

e := 1

f := 1

e + f6 2

e*x + f*y6 1

maximize(x + y)

declarations

x,y : mpvar

e,f : uncertain

end-declarations

e := 2

f := 5

e + f6 2

e*x + f*y6 1

maximize(x + y)

All three versions of the model solve to the same solution of x = y = 0.5;

The reasons behind this may not be immediately obvious but become clear if we write out the effect of

rule 1 on, say, the 3rd case:

Case 3 becomes which is

declarations

x,y : mpvar

e,f : uncertain

end-declarations

e := 2

f := 5

e + f6 2

e*x + f*y6 1

maximize(x + y)

declarations

x,y : mpvar

e,f : uncertain

end-declarations

e+2 + f+56 2

(e+2)*x + (f+5)*y6 1

maximize(x + y)

declarations

x,y : mpvar

e,f : uncertain

end-declarations

e + f6 -5

(e+2)*x + (f+5)*y6 1

maximize(x + y)

As the shifts are also applied to the uncertain constraints (including bounds), in the example they cancel

each other out: the linear robust constraints and the corresponding error constraints have been shifted

together.

However, there is a very real difference should the model be solved using its nominal values. To

demonstrate this, let us explicitly write out the nominal equivalent for the 3 cases:

The nominal equivalent

of case 1:

The nominal equivalent

of case 2:

The nominal equivalent

of case 3:

declarations

x,y : mpvar

e,f : uncertain

end-declarations

0*x + 0*y6 1

maximize(x + y)

declarations

x,y : mpvar

e,f : uncertain

end-declarations

1*x + 1*y6 1

maximize(x + y)

declarations

x,y : mpvar

e,f : uncertain

end-declarations

2*x + 5*y6 1

maximize(x + y)

is unbounded. solves to x=1 and y=0. solves to x=0.5 and y=0.

1.4 Examples of robustmodels
In the remainder of this whitepaper we provide several full-blown examples of robust optimization

applied to real-world problems. Themodel and data files associated with these examples are available

in the FICOR© Xpress Optimization distribution (subdirectory examples/robust/Mosel).

2 Robust shortest path

2.1 Problem description
Every day you are driving from your home to your work location. There are a number of alternative

routes that you might use. Youmay think of these routes as a set of lines that connect nodes

(=intersection points of routes). For every line (edge) we know the required travel time for getting from

one end point to the other. What we don’t know is the occurrence of roadworks or similar incidents that

slow down traffic and hence cause delays.

8

Figure 1: Road network

Figure 1 shows an example of a road network where the journey start and end points are marked as

’Source’ and ’Sink’ respectively. The label tuples (L,D) on the edges indicate the length (travel time) of an

edge and the maximum delay D that may result from roadwork on this edge. In this data instance, the

times in both directions for traveling along an edge are the same and the maximum delay in both senses

is also the same, but in the general case the values might actually be different.

2.2 Mathematical formulation

2.2.1 Shortest path problem

The problem of finding the least cost route through a network is known as the shortest path problem.

We can state this problem via binary variables usea associated with each arc awhere usea = 1 if arc a is
selected and 0 otherwise.

For every node in the network except the source and sink nodes we establish a flow balance constraint:

the sum of incoming travel is the same as the outgoing travel. The source node only has a single

outgoing used arc, the sink node has a single incoming used arc. In the following, we define an arc a via

its originARCa,1 and its destination nodeARCa,2.

min
∑

a∈Arcs

LENa · usea

s.t.
∑

a∈Arcs|ARCa,1=Source

usea = 1,
∑

a∈Arcs|ARCa,2=Sink

usea = 1

∑
a∈Arcs|ARCa,2=Source

usea = 0,
∑

a∈Arcs|ARCa,1=Sink

usea = 0

∀n ∈ Nodes− {Source,Sink} :
∑

a∈Arcs|ARCa,2=n

usea =
∑

a∈Arcs|ARCa,1=n

usea

∀a ∈ Arcs : usea ∈ {0, 1}

In themodel formulation above, we can add the delays caused by roadworks as an additional coefficient

to the objective function so that they are taken into account by the shortest path calculation:

min
∑

a∈Arcs

(LENa +MAXDELAYa) · usea

However, this means that we assume

(a) all delays take their maximum value and

(b) all roads incur delays.

These assumptions yield an overly conservative prediction. This case provides an upper bound on the

total travel time estimate but it can certainly be considered as fairly unlikely to happen.

2.2.2 Robust optimization problem

The problem that we really wanted to state is

(a) delays take values up to their specifiedmaximum value and

(b) up toN roads incur delays.

This means that the value of the delay per arc is not fixed, but uncertain. If delaya denotes the

uncertain duration of the delay associates with arc a ∈ Arcswe can state the cardinality constrained
uncertainty set as follows:

U = {delay : |{delaya : delaya > 0}| 6 N, 0 6 delaya 6 MAXDELAYa∀a ∈ Arcs}

The resulting robust optimization problem then has the following form.

min
∑

a∈Arcs

(LENa + delaya) · usea

s.t.
∑

a∈Arcs|ARCa,1=Source

usea = 1,
∑

a∈Arcs|ARCa,2=Sink

usea = 1

∑
a∈Arcs|ARCa,2=Source

usea = 0,
∑

a∈Arcs|ARCa,1=Sink

usea = 0

∀n ∈ Nodes− {Source,Sink} :
∑

a∈Arcs|ARCa,2=n

usea =
∑

a∈Arcs|ARCa,1=n

usea

∀a ∈ Arcs : usea ∈ {0, 1}

delay ∈ U = {delay : |{delaya : delaya > 0}| 6 N, 0 6 delaya 6 MAXDELAYa∀a ∈ Arcs}

2.3 Implementation
The following Mosel model roadworks.mos implements the robust optimization model from the

previous section. Notice the use of cardinality on the uncertains delay to limit the total number of

occurrences of roadworks to at most NWORK (the parameter forN) with a default value of 2. The

uncertains that are used in such a cardinality constraint need to have explicit lower and upper
bounds set. The solution reporting displays the selected arcs and the values of the uncertains for which

a nonzero delay is chosen by the opponent.

model "road network"
uses "mmxprs", "mmrobust"

parameters
DATAFILE="roads_9.dat"
NWORK = 2 ! Number of roadworks

end-parameters

declarations
Nodes: range ! Set of nodes
ARC: array(Arcs:range,1..2) of integer ! Arc origins/destinations
LEN,MAXDELAY: array(Arcs) of real ! Length and max delay per arc
Source,Sink: integer ! Source and sink node numbers

use: array(Arcs) of mpvar ! 1 iff arc is used
TotalLength: robctr ! Objective function
delay: array(Arcs) of uncertain ! Uncertain delay

end-declarations

!**** Input datafile ****
initializations from DATAFILE

Nodes Arcs Source Sink ARC
[LEN,MAXDELAY] as "ArcData"

end-initializations

!**** Robust problem formulation ****
forall(a in Arcs) use(a) is_binary

! Sink and source of flow
sum(a in Arcs | ARC(a,1)=Source) use(a)=1
sum(a in Arcs | ARC(a,2)=Sink) use(a)=1
sum(a in Arcs | ARC(a,2)=Source) use(a)=0
sum(a in Arcs | ARC(a,1)=Sink) use(a)=0

! Flow balance in intermediate nodes
forall(n in Nodes-{Source,Sink})

sum(a in Arcs | ARC(a,2)=n) use(a) = sum(a in Arcs | ARC(a,1)=n) use(a)

! Random construction work on NWORK arcs
forall(a in Arcs) delay(a) <= MAXDELAY(a)
forall(a in Arcs) delay(a) >= 0
cardinality(union(a in Arcs) {delay(a)}, NWORK)

! Shortest path length
TotalLength:= sum(a in Arcs) (LEN(a)+delay(a))*use(a)

!**** Solving ****
minimize(TotalLength)

! Solution reporting
if getprobstat=XPRS_OPT then

writeln("Robust shortest path: ", getobjval)
writeln("Delay: ", sum(a in Arcs | use(a).sol>0) delay(a).sol)
forall(a in Arcs | use(a).sol>0)
writeln(if(ARC(a,1)=Source, "Source", string(ARC(a,1))), " -> ",

if(ARC(a,2)=Sink, "Sink", string(ARC(a,2))), "; ")
forall(a in Arcs | delay(a).sol>0)
writeln("Delay on (", ARC(a,1), ",", ARC(a,2), "): ", delay(a).sol)

else
writeln("No solution")

end-if

end-model

2.4 Results
Figure 2 shows the path through the network defined by the robust solution; it has a total length of 21.

The path it follows, i.e., Source→ 2→ 5→ 6→ 7→ Sink, has a nominal length of 14, but if roads 2–5 and

5–6 have construction work wemust add a total delay of 7. Given that these two roads havemaximum

delay (together with the combination of roads 2–5 and 6–7), any other combination of at most two

roads with construction will result in a total duration of 21 minutes or less. The delay that wemust add to

the nominal length of the path is given by the sum of the solution values of the uncertains delay(a)
associated with every arc a in the path. The solution values of the uncertains delay can be interpreted
as providing an answer to the question ’which is the worst possible choice of delays for the selected

path?’. The resulting values of the uncertains place the delay on the two arcs (2,5) and (5,6) for a total

delay of 7.

Delay on (2,5): 4
Delay on (5,6): 3

Figure 2: Robust solution with 2 delays

Consider now the solution in Figure 3, obtained by ignoring the delays caused by road work. The optimal

path in this case is Source→ 3→ 8→ Sink, with a total duration of 11. However, this is a very optimistic

solution as in the worst case the roads Sink–3 and 3–8may have construction work, in which case the

travel time increases by 13 minutes to 24, clearly worse than the robust solution of 21 minutes.

Figure 3: Zero-delay solution

Finally, let us take a look at the solution obtained by setting all arc costs to their maximum value, i.e., a

shortest path computed on a graph where all edges are assumed to have their maximum delay. As

shown in Figure 4, the solution is the path Source→ 2→ 4→ 7→ Sink, and has a duration of 27 minutes.

Under the assumption that at most two roads have construction work, this travel time is meaningless: in

fact, the nominal travel time is 13 minutes and the worst-case travel time is obtained with road work on

2–4 and 4–7, which add 10minutes to the solution and hence yield a total travel time of 23, still worse

than the robust shortest path of 21 depicted above.

Figure 4: Maximum-delay solution

We recap the above results in Table 1, which shows the worst-case travel times for the paths determined

by the three approaches (robust for N=2, shortest path with nominal values, shortest path where all

roads havemaximum delay) in five cases: no road work, road work inN = 1, 2, or 3 edges of the graph,

and road work everywhere, i.e. N is infinite.

Table 1: Path costs

N 0 1 2 3 +inf Path

Robust 14 18 21 24 28 Source→ 2→ 5→ 6→ 7→ Sink

Nominal 11 18 24 30 30 Source→ 3→ 8→ Sink

All delays 13 19 23 25 27 Source→ 2→ 4→ 7→ Sink

The values in this table can be obtained by fixing the arc selection in the robust problem to a given path

and re-solving it for different values ofN (i.e. we determine the worst-case delay for a given path that

satisfies all constraints of the robust problem).

Note that, under the assumption we havemade at the beginning of this section, the only solution that

truly solves the problemwith minimumworst-case travel time is the one that was found using the

uncertains to model our knowledge, albeit limited, of the road works in this network.

3 Production planning under demand uncertainty

3.1 Problem description
As the manager of a plant producing drinking glasses you wish to plan the production for the next 12

weeks. Your plant produces six different types (V1 to V6) in batches of 1000 glasses; batches may be

incomplete (fewer than 1000 glasses). For every glass type you have got demand estimate for the 12

coming weeks. The initial stock and as well as the required final stock level (in thousands) are known. Per

batch of every glass type, the production and storage costs inBC are given, together with the required

working time for workers andmachines (in hours), and the required storage space (measured in

numbers of trays).

3.2 Mathematical formulation

3.2.1 Multi-period, multi-item production planning problem

Let PRODS be the set of products (glass types) andWEEKS = {1, ...,NT } the set of time periods. We

writeDEMpt for the demand for product p in time period t. We also haveCPRODp andCSTOCKp the

production and storage cost for glass type p. This cost is identical for all time periods, but it would be

easy to model a different cost per time period by adding an index for the time period.

TIMEWp and TIMEMp denote the worker andmachine times respectively required per unit of product

p, and correspondingly, SPACEp the storage area. The initial stock ISTOCKp is given, as is the desired

final stock level FSTOCKp per product (see data in Table 2). We writeCAPW andCAPM for the

capacities of workers andmachines respectively, andCAPS for the capacity of the storage area.

To solve this problem, we need variables producept to represent the production of glass type p in time

period t. The variables corresponding to the stock level of every product p at the end of period t are
called storept. By convention, the initial stock level ISTOCKp may be considered as the stock level at the

end of time period 0 and we use the notation storep0 to simplify the formulation of the stock balance

constraints:

∀p ∈ PRODS, t ∈ WEEKS : storept = storep,t−1 + producept −DEMpt

These stock balance constraints state that the quantity storept of product that is held in stock at the
end of a time period t equals the stock level storep,t−1 at the end of the preceding period plus the

production producept of the time period tminus the demandDEMpt of this time period.

We wish to have a certain amount of product in stock at the end of the planning period to avoid that

stocks run down to zero at the end of the planning horizon. These constraints on the final stock levels are

expressed by the constraints:

∀p ∈ PRODS : storep,NT > FSTOCKp

We now formulate the various capacity constraints for every time period. The following constraints

guarantee that the capacity limits on manpower, machine time, and storage space are kept:

∀t ∈ WEEKS :
∑

p∈PRODS

TIMEWp · producept 6 CAPW

∀t ∈ WEEKS :
∑

p∈PRODS

TIMEMp · producept 6 CAPM

∀t ∈ WEEKS :
∑

p∈PRODS

SPACEp · producept 6 CAPS

The cost function that is to be minimized is the sum of production and storage costs for all products and

time periods.

min
∑

p∈PRODS

∑
t∈WEEKS

(CPRODp · producept + CSTOREp · storept)

We obtain the complete mathematical model by the non-negativity constraints for the production

variables and for the stored quantities to the constraints described above.

min
∑

p∈PRODS

∑
t∈WEEKS

(CPRODp · producept + CSTOREp · storept)

s.t. ∀p ∈ PRODS, t ∈ WEEKS : storept = storep,t−1 + producept −DEMpt

∀p ∈ PRODS : storep,NT > FSTOCKp

∀t ∈ WEEKS :
∑

p∈PRODS

TIMEWp · producept 6 CAPW

∀t ∈ WEEKS :
∑

p∈PRODS

TIMEMp · producept 6 CAPM

∀t ∈ WEEKS :
∑

p∈PRODS

SPACEp · producept 6 CAPS

∀p ∈ PRODS, t ∈ WEEKS : producep,t > 0

∀p ∈ PRODS, t ∈ WEEKS : storep,t > 0

3.2.2 Robust optimization problem

Various assumptions behind the mathematical model that we have stated abovemight be questioned:

1. Constant resource capacity: availability of personnel most likely will not be constant over time

(subject to holidays, training, sick leave etc.) and there is also a risk of scheduled (maintenance) or

unscheduled (breakdown) machine outages.

2. Exact demand quantities are known: demand forecasts typically are estimates, most often

resulting from an analysis of historical values.

Workers’ absence

The case of machine outage (formulated as k contingencies) is studied in Sections 7 and 6 of this

whitepaper. Let us therefore here take a look at howwemight capture the uncertainty in the availability

of personnel.

LetABSENCEt be amaximum limit on the absence hours per time period t (the actual absence will take
at most this value), and we also know by experience what is the average absence over a longer period of

time (expressed as a percentageA of the total worker hours). We introduce uncertains absentt to
represent the actual absence hours per time period. The worker capacity constraints from the original

model are then replaced by the following:

∀t ∈ WEEKS :
∑

p∈PRODS

TIMEWp · producept 6 CAPW − absentt

absent ∈ Uabsent

where the polyhedral uncertainty setUabsent is characterized by

Uabsent = {absent :
∑

t∈WEEKS

absentt 6 A · CAPW · |WEEKS|,

0 6 absentt 6 ABSENCEt∀t ∈ WEEKS}

Demand scenarios

For a robust formulation of the demand, assume that we have got a number of different

scenarios—obtained from historical data for comparable periods or possibly resulting from different

forecasting methodologies—that describe the space of possible realizations of demand. In the place of

the fixed demand quantitiesDEMpt we nowwork with uncertain quantities demandpt that are

determined by the demand scenario data SCENDEMspt for a given set of scenarios s ∈ SCEN. The

demand s included in the original model formulation via the stock balance constraints.

∀p ∈ PRODS, t ∈ WEEKS : storept = storep,t−1 + producept −DEMpt

Anaive ’robustification’ might attempt to simply replace the fixed demand quantities by the uncertains

demandpt. However, this approach would not lead to the desired result: by introducing a single

uncertain quantity in an equality constraint we do not leave any room for different realizations of the

uncertain. We therefore nowwork with the following two sets of inequalities in the place of the stock

balance constraints.

∀p ∈ PRODS, t ∈ WEEKS : storept 6 storep,t−1 + producept − demandpt

∀p ∈ PRODS, t ∈ WEEKS : storept > storep,t−1 + producept −maxs∈SCENSCENDEMpt

The first of these inequalities states that the demandmust be satisfied from the production in a period

and the difference between stock levels at the beginning and end of the period, which can bemore

easily seen in this transformed version of the inequality:

demandpt 6 storep,t−1 + producept − storept

The second inequality forces the final stock per period to be at least what remains after satisfying the

largest possible demand.

3.3 Implementation
The standard deterministic model can be implemented as follows.

model "C-2 Glass production"
uses "mmxprs"

declarations
NT = 12 ! Number of weeks in planning period
WEEKS = 1..NT
PRODS = 1.. 6 ! Set of products

CAPW,CAPM: integer ! Capacity of workers and machines
CAPS: integer ! Storage capacity
DEM: array(PRODS,WEEKS) of integer ! Demand per product and per week
CPROD: array(PRODS) of integer ! Production cost per product
CSTOCK: array(PRODS) of integer ! Storage cost per product
ISTOCK: array(PRODS) of integer ! Initial stock levels
FSTOCK: array(PRODS) of integer ! Min. final stock levels
TIMEW,TIMEM: array(PRODS) of integer ! Worker and machine time per unit
SPACE: array(PRODS) of integer ! Storage space required by products

produce: array(PRODS,WEEKS) of mpvar ! Production of products per week
store: array(PRODS,WEEKS) of mpvar ! Amount stored at end of week
end-declarations

initializations from 'c2glass.dat'
CAPW CAPM CAPS DEM CSTOCK CPROD ISTOCK FSTOCK TIMEW TIMEM SPACE
end-initializations

! Objective: sum of production and storage costs
Cost:=
sum(p in PRODS, t in WEEKS) (CPROD(p)*produce(p,t) + CSTOCK(p)*store(p,t))

! Stock balances
forall(p in PRODS, t in WEEKS)

Bal(p,t):=
store(p,t) = if(t>1, store(p,t-1), ISTOCK(p)) + produce(p,t) - DEM(p,t)

! Final stock levels
forall(p in PRODS) store(p,NT) >= FSTOCK(p)

! Capacity constraints
forall(t in WEEKS) do
LimitW(t):= sum(p in PRODS) TIMEW(p)*produce(p,t) <= CAPW ! Workers
LimitM(t):= sum(p in PRODS) TIMEM(p)*produce(p,t) <= CAPM ! Machines
LimitS(t):= sum(p in PRODS) SPACE(p)*store(p,t) <= CAPS ! Storage
end-do

! Solve the problem
minimize(Cost)

! Solution printing
writeln("Total cost: ",getobjval)
end-model

Workers’ absence

For the implementation of the more fine-grained handling of workers’ absence we introduce uncertains

absent that are bounded by the estimatedmaximum number of absence hours per time period

(ABSENCE) and we equally assume a limit of 5% on the total absence time across the planning period. In

the work capacity constraints the absence needs to be deduced from the theoretically available work

hours (we here assume that this value is 20% higher than the default limit in the basic model).

declarations
ABSENCE: array(WEEKS) of real ! Max. absence (hours)
absent: array(WEEKS) of uncertain ! Absence of personnel
end-declarations

! Limit on total absence (uncertainty set)
sum(t in WEEKS) absent(t) <= 0.05*CAPW*WEEKS.size
forall(t in WEEKS) absent(t) <= ABSENCE(t)
forall(t in WEEKS) absent(t) >= 0

! Uncertains occur in several constraints
setparam("ROBUST_UNCERTAIN_OVERLAP", true)

! Worker capacity constraints
forall(t in WEEKS)

LimitW(t):= sum(p in PRODS) TIMEW(p)*produce(p,t) <= CAPW*1.2 -absent(t)

Demand scenarios

For the formulation of the demand scenarios we need to modify the definition of the constraints that

involve demand data, that is, the stock balance constraints. Before stating the scenario constraints, we

need to copy the scenario data into the form that is expected by the scenario construct, namely the

array SCENDATA that is indexed by the scenario counter and the uncertain demand associated with

every time period.

declarations
SCEN: range ! Demand scenarios
SCENDEM: array(SCEN,PRODS,WEEKS) of integer ! Demand per product & week
demand: array(PRODS,WEEKS) of uncertain ! Uncertain demand

SCENDATA: array(SCEN,set of uncertain) of real ! Aux. data structure
end-declarations

! Demand scenarios
forall(s in SCEN, p in PRODS, t in WEEKS | SCENDEM(s,p,t)>0)

SCENDATA(s, demand(p,t)):= SCENDEM(s,p,t)
scenario(SCENDATA)

! Stock balances
forall(p in PRODS, t in WEEKS) do

Bal(p,t):=
store(p,t) <= if(t>1, store(p,t-1), ISTOCK(p)) + produce(p,t) - demand(p,t)

Bal2(p,t):=
store(p,t) >= if(t>1, store(p,t-1), ISTOCK(p)) + produce(p,t) -
max(s in SCEN) SCENDEM(s,p,t)

end-do

! Uncertains occur in several constraints
setparam("ROBUST_UNCERTAIN_OVERLAP", true)

Both sets of uncertainties can be applied at the same time. However, for the analysis of their effect it

might be preferrable to apply only one at a time.

3.4 Results
The original problem description and instance data are taken from [?] (Section 8.2 Production of

drinking glasses). The costs and resource usage data for the six glass types are listed in Table 2.

Table 2: Data for six glass types

Production cost Storage cost Initial stock Final stock Timeworker Timemachine Storage space

V1 100 25 50 10 3 2 4

V2 80 28 20 10 3 1 5

V3 110 25 0 10 3 4 5

V4 90 27 15 10 2 8 6

V5 200 10 0 10 4 11 4

V6 140 20 10 10 4 9 9

The solution of the original problemwith the basic demand scenario has a total cost ofBC 185,899. The

resulting detailed production plan is displayed in Table 3. We find that the available manpower is fully

usedmost of the time (in the first week, 351 hours are worked, in all other weeks the limit of 390 hours is

reached) and the machines are used to their full capacity in certain time periods (weeks 1-3 and 5), but

the available storage space is in excess of the actual needs.

If we try to introduce robustness to such a tightly constrained problem, the outcomemost likely will be

’problem infeasible’ as there is no slack to incorporate alternatives. We have therefore somewhat

relaxed the original bound on the personnel hours, assuming that its original value represents a rough

worst-case estimate and that the actually available working hours are 20% higher. The more

fine-grained estimate of absence results in the solution summarized in Table 4. The overall cost ofBC
181,210 is slightly lower than the previous, fixed worst-case absence case.

The scenario-based approach results in a higher total cost ofBC203,545. In the summary results in Table

5 we observe that both, machine capacity and workers’ hours, are most of the time used at or close to

their capacity limit (850 and 429=1.1 · 390 respectively). Furthermore, larger quantities of products are

held in stock.

For easier comparison, Table 6 shows the summary results for the original model with the same capacity

limits as have been used wit the robust formulation in Table 5, the total cost is in this caseBC 181,432.

Table 3: Optimal production plan for basic demand scenario

Week 1 2 3 4 5 6 7 8 9 10 11 12

V1 Prod. 8.8 5.5 0.6 30.2 27.4 8.6 23 20 29 30 28 42

Store 38.8 22.2 4.8 - 10.4 – – – – – – 10

V2 Prod. 0 16 23 20 11 10 12 34 21 23 30 22

Store 3 – – – – – – – – – – 10

V3 Prod. 18 35 17 10 9 21 23 15 10 0 13 27

Store – – – – – – – – – – – 10

V4 Prod. 16 45 24 38 41 20 19 37 28 12 30 47

Store – – – – – – – – – – – 10

V5 Prod. 47.7 14.6 35.1 14.3 23.5 22.8 43.8 0 26.5 2.8 0 0

Store 24.7 19.3 31.4 30.8 44.2 45 70.8 40.75 39.25 35 20 10

V6 Prod. 12 18 20 19 18 35 0.8 27.2 12 49 29.2 5.8

Store – – – – – – 0.75 – – 19 27.25 10

Workers 351 390 390 390 390 390 390 390 390 390 390 390

Machines 850 850 850 753.2 850 836.8 790 675.2 742.5 650.2 641.2 641.8

Space 268.8 166.2 144.8 123 218.4 180 289.8 163 157 311 325.2 330

Table 4: Summary results with robust formulation of absence

Week 1 2 3 4 5 6 7 8 9 10 11 12

Workers 243 378 370 407 325 398 383 414 412 445 429 437

CAPW-ABSENCE 437 437 429 429 414 398 398 414 445 445 429 437

Machines 609 850 736 770 736 794 772.2 739.8 803 839.5 734 747.5

Space 152.5 32 0 0 20 0 99 0 16 130 219.5 330

Table 5: Summary results with demand scenarios

Week 1 2 3 4 5 6 7 8 9 10 11 12

Workers 363 425 425 425 425 425 425 425 425 425 425 425

Machines 850 850 850 850 850 850 850 850 850 786.7 711 700.3

Space 230.7 145.9 150.4 126.3 233.2 211.9 306.8 181.7 169.7 256.7 305.7 330

Table 6: Summary results for deterministic model with 1.1 · CAPW

Week 1 2 3 4 5 6 7 8 9 10 11 12

Workers 243 378 370 407 305 418 393 425 425 425 425 425

Machines 609 850 736 770 681 849 799 771.2 840 783.2 721 721.5

Space 152.5 32 0 0 0 0 108.7 21.2 50.6 151.5 245.5 330

4 Robust network design

4.1 Problem description
Suppose we want to design a Telecommunication network. We are given a set of routers, a set of links,

and a set of traffic demands (measured in MB/s) between pairs of routers; for convenience we can

assume each of them sits in a city. We want to find the capacity to be installed on each link in order to

satisfy all traffic demands.

There are several variants of this problem in the scientific literature, and an important subclass deals

with the problem of provisioning virtual private networks, or VPNs for short.

In general, in VPN traffic demand is difficult to predict even in the short term, and hence the traffic

demands are uncertain; however, there exist uncertainty models for such traffic demands. Here we

consider one that has attracted considerable attention and that is realistic enough in VPNs.

We suppose that there are two parameters s+i and s−i which estimate the maximum total outgoing and

incoming traffic, respectively, at node i.

The problem is hence that of finding the value of the capacity of every arc of the network (which is a

multiple of a given unit capacityC) so as to accommodate any traffic demand dhk from node h to node

k, subject to uncertainty on these demands modeled by the upper bounds on the incoming and outgoing

traffic for each node of the network.

4.2 Mathematical formulation
Consider a directed graphGwith n nodes andm arcs, defined asG = (V ,A), where V = {1, ...,n}
denotes the set of n nodes andA = {(i1, j1), (i2, j2), ..., (im, jm)} is the set of arcs. The cost of every unit
of capacity (MB/s) installed at each arc (i, j) ∈ A is denoted by aij, and, for each pair of nodes (h,k) the
(unknown) traffic demand from h to k is given by dhk.

Let us introduce a binary flow variable fhk
ij : this variable is one if demand (h,k) is routed through arc

(i, j), zero otherwise, and serves to model the flow of data between each source/destination pair. We

also need to introduce the integer variable xij, which represents the number of channels of capacityC
to be installed in the network.

This problem is a particular case of amulti-commodity network flow problems, with the important

addition of the uncertain in the traffic demands.

Let us first model the uncertainty set: every uncertain demand dhk is nonnegative and the total demand

leaving or entering a node i is bounded from above by a given parameter:

∀k ∈ V ,
∑

h∈V :h 6=k

dhk 6 s−k

∀h ∈ V ,
∑

k∈V :k6=h

dhk 6 s+h .

Let us consider now the twomain classes of constraints for the problem: conservation of flow for the

variables f and the capacity constraints. The former reads as follows: for each node i and each
demand (h,k), if i is an intermediate node for this demand, i.e., it is neither h nor k, then the incoming

flowmust be equal to the outgoing flow, therefore

∀i,h,k ∈ V : h 6= i 6= k,
∑

j:(i,j)∈A

fhk
ij =

∑
j:(j,i)∈A

fhk
ji .

If i = h, the total balance of flowmust be one, hence

∀i,h,k ∈ V : i = h,
∑

j:(i,j)∈A

fhk
ij −

∑
j:(j,i)∈A

fhk
ji = 1.

The case for i = k is redundant and its corresponding constraint can be omitted.

We can nowwrite the capacity constraint: the total traffic on arc (i, j), weighted by the (unknown) traffic
demands, must be less than or equal to the total capacity installed on that arc:

∀(i, j) ∈ A,
∑
h∈V

∑
k∈V :k6=h

dhkf
hk
ij 6 Cxij.

Finally, the objective function, to be minimized, is

∑
(i,j)∈A

aijxij.

4.3 Implementation
Below is the implementation in Mosel: note that the option ROBUST_OVERLAP_UNCERTAIN is used since
the capacity constraints at each arc use a subset of the uncertain demands.

model "Robust Network"
uses "mmrobust", "mmxprs"

parameters
vpn_data="vpn_data.dat"

end-parameters

declarations
NODES: range ! Set of nodes
ARCCOST: dynamic array(NODES, NODES) of real ! Per-unit cost of arc (i,j)

DEM_IN: array(NODES) of real ! Total INCOMING demand at each node
DEM_OUT: array(NODES) of real ! Total OUTGOING demand at each node

UNITCAP: integer ! Per-unit capacity (independent of arc)

! Decision variables
flow: dynamic array(NODES, NODES, NODES, NODES) of mpvar

! flow(i,j,h,k) is 1 iff demand h->k uses arc (i,j)
capacity: dynamic array(NODES, NODES) of mpvar ! Capacity to install on arc (i,j)

! Uncertain parameters
demand: array(NODES, NODES) of uncertain ! Uncertain traffic demand
RobustCap: array(NODES, NODES) of robustctr ! Uncertain traffic demand, formulated at each arc

end-declarations

! The following option is necessary to allow uncertain demands to be
! used across multiple capacity constraints

setparam("robust_uncertain_overlap", true)

! Set verbosity level

setparam("xprs_verbose", true)

!**** Data input ****

initializations from vpn_data
NODES ARCCOST
DEM_IN DEM_OUT
UNITCAP

end-initializations

!**** Model formulation ****

forall(i in NODES, j in NODES, h in NODES, k in NODES |
exists(ARCCOST(i,j)) and ARCCOST(i,j) > 0 and h <> k) do

create(flow(i,j,h,k))
flow(i,j,h,k) is_binary

end-do

forall(i in NODES, j in NODES | exists(ARCCOST(i,j)) and ARCCOST(i,j) > 0) do
create(capacity(i,j))
capacity(i,j) is_integer

end-do

! Flow balance at intermediate nodes for each demand(h,k)
forall(i in NODES, h in NODES, k in NODES | i <> h and i <> k and k <> h)
sum(j in NODES | exists(flow(i,j,h,k))) flow(i,j,h,k) -
sum(j in NODES | exists(flow(j,i,h,k))) flow(j,i,h,k) = 0

! Flow balance at source nodes (unnecessary for sink nodes)
forall(i in NODES, h=i, k in NODES | k <> h)
sum(j in NODES | exists(flow(i,j,h,k)) ) flow(i,j,h,k) -
sum(j in NODES | exists(flow(j,i,h,k)) ) flow(j,i,h,k) = 1

! Capacity (robust) constraint

forall(i in NODES, j in NODES | exists(ARCCOST(i,j)) and ARCCOST(i,j) > 0)
RobustCap(i,j) :=
sum(h in NODES, k in NODES | h <> k) demand(h,k) * flow(i,j,h,k) <=

UNITCAP * capacity(i,j)

! Uncertainty set: all demands are nonnegative and constrained by
! total outgoing and incoming demand

forall(h in NODES, k in NODES | h <> k) demand(h,k) >= 0

forall(h in NODES) sum(k in NODES | h <> k) demand(h,k) <= DEM_OUT(h)
forall(h in NODES) sum(k in NODES | h <> k) demand(k,h) <= DEM_IN(h)

! Shortest path length
NetCost := sum(i in NODES, j in NODES | exists(ARCCOST(i,j)) and ARCCOST(i,j) > 0)
ARCCOST(i,j) * capacity(i,j)

!**** Solving ****

minimize(NetCost)

declarations
curNode: integer ! Local variable used in following the path of each demand

end-declarations

! Printing capacities
writeln("Robust solution has total cost ", getobjval)
forall(i in NODES, j in NODES | exists(capacity(i,j)) and capacity(i,j).sol > 0)
writeln("arc ", i, " -- ", j, ": ", capacity(i,j).sol, " units")

writeln("Paths:")

forall(h in NODES, k in NODES | h <> k) do
write("Demand ", h, " --> ", k, ": ")
curNode := h
write(h)

while(curNode <> k) do
forall(j in NODES | flow(curNode, j, h, k).sol > 0.5) do

write(" -- ", j)
curNode := j

end-do
end-do
writeln

end-do

writeln("Active demands at each arc:")

forall(i in NODES, j in NODES | exists(ARCCOST(i,j)) and ARCCOST(i,j) > 0 and
sum(h in NODES, k in NODES) getsol(demand(h,k), RobustCap(i,j)) > 0) do

write("Arc (", i, ",", j, "): ")

forall(h in NODES, k in NODES | getsol(demand(h,k), RobustCap(i,j)) > 0)
write(h, "-->", k, " (", getsol(demand(h,k), RobustCap(i,j)), "); ")

writeln
end-do

end-model

4.4 Input Data
Consider the network depicted in Figure 5, where each link represents two arcs in opposite directions.

Table 7 below reports the parameters specified in the data file vpn_data.dat describing the
uncertainty set, i.e., the maximum incoming and outgoing traffic demands for each node of the

network.

1

2

3

5

6

4

Figure 5: Network topology

Table 7: Maximum incoming/outgoing demand

Node Incoming Outgoing

1 120 105

2 34 95

3 35 82

4 101 102

5 78 76

6 75 140

Table 8 describes instead the arc cost for every arc (i, j) used in the network (a “–” means that no

connection exists).

Table 8: Arc costs

i/j 1 2 3 4 5 6

1 – 10 9 12 – –

2 8 – 10 – 12 –

3 10 7 – – 9 –

4 10 – – – 12 5

5 – 9 8 11 – 12

6 – – – 10 9 –

4.5 Results
For the example provided in vpn_data.dat, which describes a network of six nodes and 18 arcs, we

obtain a design of total cost 566. Table 9 reports the number of units of capacity to be installed on each

arc, while Table 10 describes the path followed by each demand from source to destination.

Table 9: Optimal arc capacity

i/j 1 2 3 4 5 6

1 – 2 2 13 – –

2 5 – 0 – 0 –

3 5 0 – – 0 –

4 10 – – – 4 4

5 – 0 0 4 – 0

6 – – – 7 0 –

Table 10: Optimal routes

Source Destination Path Source Destination Path

1 2 1→2 4 1 4→1

1 3 1→3 4 2 4→1→2

1 4 1→4 4 3 4→1→3

1 5 1→4→5 4 5 4→5

1 6 1→4→6 4 6 4→6

2 1 2→1 5 1 5→4→1

2 3 2→1→3 5 2 5→4→1→2

2 4 2→1→4 5 3 5→4→1→3

2 5 2→1→4→5 5 4 5→4

2 6 2→1→4→6 5 6 5→4→6

3 1 3→1 6 1 6→4→1

3 2 3→1→2 6 2 6→4→1→2

3 4 3→1→4 6 3 6→4→1→3

3 5 3→1→4→5 6 4 6→4

3 6 3→1→4→6 6 5 6→4→5

The last part of the output shows, for each arc, the demands for which the uncertain values are non-zero

in the worst case. Note that arcs such as (2, 3) do not admit any nonzero demand simply because they

are not used by the solution—the opponent has no incentive to increase the demand on such arcs.

Active demands at each arc:
Arc (1,2): 1-->2 (34);
Arc (1,3): 1-->3 (35);
Arc (1,4): 1-->4 (101); 1-->5 (4); 2-->6 (67); 3-->5 (74); 3-->6 (8);
Arc (2,1): 2-->1 (95);
Arc (3,1): 3-->1 (82);
Arc (4,1): 4-->1 (102); 5-->1 (18); 6-->2 (34); 6-->3 (35);
Arc (4,5): 1-->5 (78);
Arc (4,6): 1-->6 (75);
Arc (5,4): 5-->1 (76);
Arc (6,4): 6-->1 (106); 6-->2 (34);

5 Robust portfolio optimization
In this section, we present a robust optimization formulation of the single period portfolio selection

problem. In order to compare the results from the robust optimization, we also show how to quantify

the robustness of a solution using Monte-Carlo simulation.

5.1 Problem description
The single-period portfolio selection problem is about selecting assets from a given list in order to

create a portfolio that has the greatest expected value. The assets are bought at the market price and

their value is subject to variation. The investment budget is fixed, and each selected asset is bought

using a percentage of the available budget.

The market price for each asset is known at the time of buying and corresponds to the payment to be

made by the trader to buy one unit. The future asset value is not known, but we assume the distribution

characteristics of the random variable is known.

The trader is risk averse and wants to have some guarantees about the worst case value of the

portfolio. She considers that in the worst case the downward variation of an asset value reaches 1.5

times the variance of the asset. In other words, the trader assumes it is very unlikely that the value of an

asset decreases by more than 1.5 times its variance.

Example: Let’s assume that asset #20 has amarket price of 100$ and its variability is +/- 10$. Then the

considered worst case value of the asset is 85$.

In this context, the objective of the trader is to maximize the expected value of her portfolio, but without

taking toomuch risk. A conservative trader would definitely spend all her budget in the asset with the

highest worst case value in order to be highly protected against the worst possible outcome.

Unfortunately, this strategy would dramatically reduce the expected value of the portfolio. Another

extreme policy would be an opportunistic trader who spends all her budget in the asset with the highest

expected value, without paying much attention to the worst case realization.

Both approaches are extreme and do not take into account the fact that it is quite unlikely to see all

asset values going down (or going up) to the same degree. In practice traders often aim for portfolios

with a guaranteed probability of having a value greater than aminimum target value. This is also known

as chance constrained optimization. In the ’Results’ section (5.4) we will show how robust optimization

can be used to provide solutions with a large expected value without sacrificing protection against worst

case realization.

5.2 Mathematical formulation
The set S contains the available assets, and the index s ∈ S represents an asset from this set. The

market price of the asset s is noted PRICEs, and its variability range is given by VARs.

For each asset s ∈ S, the variable xs represents the fraction of the budget spent to buy asset s. The
variablew is used to represent the worst case value of the portfolio. The variable devs describes the
possible variation of the value of the asset s.

The valueN describes the worst case the trader will consider and corresponds to the maximum

decrease of the asset value expressed as amultiple of its standard deviation.

5.2.1 Highest protection

A conservative trader will want to maximize the worst case value of her portfolio. If she wants to get the

best protection against worst case realization, then she will try to maximize the worst case value of her

portfolio. In our example, this worst case realizes when the value of every asset decreases byN times its

standard deviation.

max w

s.t. w 6
∑
s∈S

(PRICEs +N · devs) · xs (with devs = −VARs)∑
s∈S

xs = 1

0 6 xs 6 1

5.2.2 Budgeted protection

As wewill see in the discussion of the results (Section 5.4), achieving a high protection against worst case

realization incurs an important loss of value in the average case. And hence, this strategy may not be

such a clever choice because average cases are more likely to happen. The trader therefore might want

to improve her model by adding a control parameter to limit the protection level.

The overall variation budgetG is defined by the following formula:

G =
∑
s∈S

(N · VARs)
2

The ellipsoidal uncertainty set

Let k be the percentage value representing the protection level. The possible asset value decrease is
controlled by the uncertainty setU(k) that is defined as follows:

U(k) = {e :
∑
s∈S

e2s 6 (k ·G)2}

Robust constraints

The resulting robust optimization problem can then be stated as:

max w

s.t. w 6
∑
s∈S

(PRICEs + es) · xs (∀e ∈ U(k))∑
s∈S

xs = 1

0 6 xs 6 1

5.3 Implementation
The following Mosel model implements the mathematical model from the previous section. Notice how

the ellipsoid uncertainty set is added to the original problem to create a robust optimization problem.

Also, take a look at how the Monte-Carlo simulation method is applied to quantify the quality of the

solution: inNMC = 5000 iterations we draw random values for the expected value of every asset by

applying a normal distribution centered around its price (mean value) and using its known variance. We

count the occurrence of objective function values (or more precisely, which value range the resulting

solution value belongs to) to determine the probabilities of different solution qualities. This method is

particularly helpful to gain some insight about the solution quality when it is difficult to determine the

exact shape of the distribution function of a random variable.

model "Robust portfolio optimization"
uses "mmrobust", "random"
uses "mmsystem"

parameters
ZEROTOL=1e-6
SEED=12345
NMC=500000
N=1.5 ! Worst case metric

end-parameters

declarations
Problems: set of mpproblem
ProtectLevel: set of real

mp_problemA: mpproblem ! Worst case optimization
mp_problemB: array(ProtectLevel) of mpproblem ! Robust optimization
NSHARES = 25 ! Max number of shares
Shares = 1..NSHARES ! Set of shares

PRICE: array(Shares) of real ! Expected return on investment
VAR: array(Shares) of real ! Uncertainty measure (deviation) per SHARE

expReturn: mpvar ! Expected portfolio value
wstReturn: mpvar ! Worst case portfolio value
frac: array(Shares) of mpvar ! Fraction of capital used per share
frac_sol: array(Shares,Problems) of real
obj: mpvar

e: array(Shares) of uncertain ! Deviation of share values
PRICEthr: array(Problems,Shares) of real ! The price threshold

end-declarations

!***************************Configuration*************************
setparam("XPRS_LOADNAMES",true)

!***************************Subroutines***************************
!**** Create the nominal model ****
procedure create_nominalmodel
! Nominal model
expReturn = (sum(s in Shares) PRICE(s)*frac(s))
wstReturn = sum(s in Shares) ( PRICE(s) - N*VAR(s) )*frac(s)

! Spend all the budget
sum(s in Shares) frac(s) = 1

end-procedure

!**** Optimize for the worst case realization ****
procedure solve_det
obj = wstReturn
maximize(obj)

end-procedure

!**** Optimize for a given protection level ****
procedure solve_rob(k: real)
! The value variation domain

assert(k>=0 and k<=1)
sum(s in Shares) (e(s) / (N*VAR(s)))^2 <= (k)^2

! The robust constraint

obj <= sum(s in Shares) (PRICE(s) + e(s)) * frac(s)

maximize(obj)
end-procedure

!***************************Main model***************************
!**** Input data generation ****
setmtrandseed(12345)
cnt := 0
forall(s in Shares, cnt as counter) do
PRICE(s) := round((100*(NSHARES-cnt+1)/NSHARES))
c := (1+sqrt((NSHARES-s)/NSHARES))/3
VAR(s) := round(PRICE(s)*c*100)/100

end-do

writeln("\n *** Input Data *** ")
writeln
writeln("Shares | Mean | S.Dev. | Worst value")
forall(s in Shares)
writeln(strfmt(s,6), " |", strfmt(PRICE(s),5), " |", strfmt(VAR(s),7,2),
" |", strfmt(PRICE(s)-N*VAR(s),6,3))

writeln

!**** Optimize the worst case ****
with mp_problemA do
create_nominalmodel
solve_det
forall(s in Shares) frac_sol(s,mp_problemA) := frac(s).sol*100
! Price set by opponent is worst case price
forall(s in Shares) PRICEthr(mp_problemA,s) := PRICE(s) - N*VAR(s)

end-do

!**** Optimize the 'budgeted' worst case ****
Ks := [0.0, ! No variation on average

0.01, ! +/- 1% variation on the list
0.10, ! +/- 10% variation on the list
0.25, ! +/- 25%
1.0] ! +/-100% variation on the list

forall(k in Ks) do
create(mp_problemB(k))
with mp_problemB(k) do
create_nominalmodel
solve_rob(k)
forall(s in Shares) frac_sol(s,mp_problemB(k)) := frac(s).sol*100
forall(s in Shares) PRICEthr(mp_problemB(k),s) := (PRICE(s) + getsol(e(s)))

end-do
end-do

!**** Print results ****
writeln("\n *** Portfolio allocation results *** ")
write("\n | protection level")
write("\nShares | Wst price | Price | A |")
forall(k in Ks) write(" ", strfmt(k*100,3), "% |" ) ; writeln
forall(s in Shares) do
write(strfmt(s,6), " |", strfmt(PRICE(s)-N*VAR(s),10,0), " |",
strfmt(PRICE(s),6), " | ")

forall(mp in Problems) do
if (frac_sol(s,mp)>1) then

write(strfmt(frac_sol(s,mp),3,0), "% | ")
else

write(" | ")
end-if

end-do
writeln

end-do

writeln("\n *** Worst case price *** ")

write("\n | protection level")
write("\nShares | Wst price | Price | A |")
forall(k in Ks) write(" ", strfmt(k*100,3), "% |" ) ; writeln
forall(s in Shares) do
write(strfmt(s,6), " |", strfmt(PRICE(s)-N*VAR(s),10,0), " |",
strfmt(PRICE(s),6), " | ")

forall(mp in Problems) do
if (frac_sol(s,mp)>1) then

write(strfmt(PRICEthr(mp,s),3,0), "* | ")
else

write(strfmt(PRICEthr(mp,s),4,0), " | ")
! write(" | ")

end-if
end-do
writeln

end-do

!**** Simulate results (Monte-Carlo simulation) ****
writeln("\nRunning Monte-Carlo simulation...")
Values := {0,20,40,60,80,100}
declarations
cntV: dynamic array(Problems,Values) of real
s1: real
s2: real

end-declarations
forall(mp in Problems) do
expRev(mp) := 0.0 ; wstRev(mp) := 0.0 ; expDev(mp) := 0.0
forall(v in Values) cntV(mp,v) := 0
c := 0.0 ! Number of draws
s1 := 0 ; s2 := 0;
forall(i in 1..NMC, c as counter) do
totalVal := 0.0 ; totalRisk := 0.0
forall(s in Shares | frac_sol(s,mp)>0) do

! draw a price
p := normal(PRICE(s),VAR(s))
! calculate share revenue
r := frac_sol(s,mp)*p/100
! calculate total revenue
totalVal += r ! summing up revenue

end-do
forall(v in Values) do

if (totalVal>v) then
cntV(mp,v) += 1

end-if
end-do
expRev(mp) += totalVal
s1 += totalVal
s2 += totalVal^2

end-do
expRev(mp) := expRev(mp) / c
expDev(mp) := sqrt(NMC*s2 - s1^2)/(NMC)
forall(v in Values) cntV(mp,v) := cntV(mp,v) / c

end-do

!**** Print simulation results ****
write("\n | protection level")
write("\n | A |")

forall(k in Ks) write(" ", strfmt(k*100,3), "% |" )
write("\n Expected value |")
forall(mp in Problems) write(strfmt(round(expRev(mp)),5), " |")
write("\n Standard deviation |")
forall(mp in Problems) write(strfmt(round(expDev(mp)),5), " |")

writeln
forall(v in Values) do
write("\n P (value>"+textfmt(v,3)+" |")

forall(mp in Problems) write(strfmt(cntV(mp,v),5,2) , " |")
end-do

end-model

5.4 Results
In order to understand how the uncertainty set impacts the solution we will solve the portfolio allocation

problem for various protection levels k and compare the proposed budget allocation against the

conservative approach.

The trader knows that she can expect a portfolio value in the range of 0 to 100. She therefore wishes to

calculate for various ranges the probability that the value of the portfolio will belong to this range. She

also wants to reach a portfolio value of 60 with high probability.

5.4.1 Input Data

Table 11 shows the asset mean value which is also the market price (Mean), the standard variation of the

value (S. Dev), and the considered worst case value withN = 1.5 (Worst value).

Table 11: Asset value distribution

Shares Mean S.Dev. Worst value

1 100 65.99 1.015

2 96 62.69 1.965

3 92 59.43 2.855

4 88 56.22 3.670

5 84 53.04 4.440

6 80 49.91 5.135

7 76 46.83 5.755

8 72 43.79 6.315

9 68 40.80 6.800

10 64 37.86 7.210

11 60 34.97 7.545

12 56 32.13 7.805

13 52 29.34 7.990

14 48 26.61 8.085

15 44 23.94 8.090

16 40 21.33 8.005

17 36 18.79 7.815

18 32 16.31 7.535

19 28 13.91 7.135

20 24 11.58 6.630

21 20 9.33 6.005

22 16 7.18 5.230

23 12 5.13 4.305

24 8 3.20 3.200

25 4 1.33 2.005

Aquick glance at the input data reveals that asset #15 would be the best choice for the conservative

trader because it has the highest worst case value, whereas asset #1 would be the preferred investment

for an optimistic trader since it has the largest best case value.

5.4.2 Analysis

In order to understand how the robust optimization behaves compared to the nominal problem, we

solve the nominal problem (A) and 5 robust optimization problems parameterized by the protection level

k (ranging from 0 to 1=100%). With a protection level of k=0 the deviation budget is also zero, and hence

the solution is a very optimistic one.

In a second step, we use a Monte-Carlo method to simulate the actual value of the assets selected by

these solutions and calculate the probability with which the portfolio value lies in the various revenue

ranges.

Table 12 presents the portfolio selection suggestion for each of these problems. For the sake of

simplicity we list only those assets that are selected in at least one solution. The conservative solution (A)

is to allocate all budget to the asset with highest worst case value. The optimistic solution (k=0.0) is to

allocate the whole budget to the asset with the highest expected value. With an increasing protection

level the suggested solutions improve the balance between assets with a high largest expected value

and those with a high worst case value.

Table 12: Results of the parameterized robust optimization

Shares Wst price Price A k=0.0% k=1% k=10% k=25% k=100%

1 1 100 100% 100% 60% 33% 11%

2 2 96 35% 28% 11%

3 3 92 5% 22% 11%

4 4 88 14% 11%

5 4 84 3% 11%

6 5 80 10%

7 6 76 10%

8 6 72 9%

9 7 68 7%

10 7 64 5%

11 8 60 2%

15 8 44 100%

Table 13 presents the results of the Monte-Carlo simulation of the portfolio value for each portfolio

selection suggestion. The conservative solution (A) is the one with highest worst case value, and as

expected the probability of this portfolio’s value to go above 60 is very low. From the trader’s point of

view this is not a good solution, even in the average case for which she will get a value of 44.

The optimistic solution (k=0%) and the solution with the lowest protection level (k=1%) have the largest

expected value. However, the trader cannot use these two solutions because she won’t be able to

commit to achieving aminimum value of 60 since the probability of obtaining a portfolio value greater

than 60 is only 73% for these two solutions.

The behavior of the other solutions (k > 10%) shows that with an increasing deviation budget, the

probability of achieving a portfolio value higher than 60 also increases.

The worst case value of the portfolio equally increases while the expected value decreases. The

volatility of the distribution of the values against the three ranges tends to reduce.

The solution with k = 100% protects the trader against important losses and allows her to commit to a

minimum portfolio value of 60 with a likelyhood of more than 90% and a total expected value of 83. A

more risk-prone trader may want to reduce the protection level to say, 10% resulting in a considerably

higher average outcome (98) and still having a 90% chance of achieving a portfolio value greater than

40.

Table 13: Results of the Monte-Carlo simulation

A k=0.0% k=1% k=10% k=25% 100%

Expected value 44 100 100 98 95 83

Standard deviation 24 66 66 45 32 17

P (value> 0) 97% 94% 93% 98% 100% 100%

P (value> 20) 84% 89% 89% 96% 99% 100%

P (value> 40) 57% 82% 82% 90% 96% 99%

P (value> 60) 25% 73% 73% 80% 86% 91%

P (value> 80) 07% 62% 62% 66% 68% 58%

P (value> 100) 01% 50% 50% 48% 44% 17%

5.5 References
The presented problem is inspired by the Single-Period Portfolio selection problem described in [?].

Applications of robust optimization to portfolio optimizations models date back to as early as 2000 [?].

6 Robust unit commitment

6.1 Problem description
The unit commitment problem consists in scheduling the power generation level of a set of generators

to meet the power demand. In order for a power system to operate safely, the power generation must

equal the demand at any time (the so-called power balance constraints) Failing to satisfy this

requirement may result in serious issues, possibly causing a black out.

The power generation level of a power generating unit depends on its commitment. When the unit is

switched off, no power is produced, only when it has been started the unit can generate electricity. Due

to technical constraints, the power generation level must lie within a band describing safe operation.

For the vast majority of unit types, the minimum power generation level is an important parameter

when deciding whether to start a particular unit or not.

Another parameter that plays an important role in the decision whether to start a unit is its startup cost.

Units with low startup cost could be started and shut downmultiple times during a planning period, while

units with high cost will preferrably be kept running once started.

For the sake of simplicity, other considerations like ramp rate constraints, shutdown costs or the various

reserve types are omitted from the present model.

Example: We shall consider the case of day-head unit commitment scheduling. This operational

planning task consists in determining on a day for the next, the startup and shutdown phases of the

generating units in order to be able to safely satisfy the next day’s power demand. However, the exact

demand is not known for sure at the time of scheduling, and hence it is considered an uncertain

parameter of the optimization problem. Common practice for dealing with this uncertainty is to commit

a larger number of units than strictly required to be sure to have sufficient upward and downward

power generation reserve. In the following we will show how to apply robust optimization techniques to

take into account this uncertainty more efficiently.

We are going to present two robust implementations of the unit commitment problem resulting in plans

for which the unit commitment decisions do not have to be changed in real time. More precisely, the

suggested unit commitment schedules make sure that enough power generation capacity is available to

supply the actual demand without requiring any last minute unit startups or shutdowns. The first model

is based on a scenario uncertainty set, it ensures that the unit commitment status will not require any

modification even if the real power demand is different from the forecast. In this problem it is assumed

that the power demand realization will be similar to either the forecast or some historical power

demand curve. The secondmodel, based on a cardinality uncertainty set, guarantees that the unit

commitment status will not require any modification even if up to ’k’ generating units are simultaneously

forced in outage.

6.1.1 Robust against power demand variation

Typical unit commitment robustness against power demand variation is achieved by adding constraints

to bound the difference between the total power generation and the total available power generation

capacity of running units. This value is known as the upward power generation level and defines the

maximum power demand increase that can be safely supplied without requiring new unit startup. This

empirical value is arbitrarly determined from historical power demand.

We will see how the robust optimization framework can be used to seamlessly implement similar

constraints by replacing the reserve requirement by a set of robust constraints formed from historical

power demand. These historical data and the power demand forecast are used to determine the real

power demand. This real power demand is modeled with the scenario uncertainty sets. The figure 6

presents the next day’s historical power demand of the last 6 years alongside with the forecasted power

demand.

Year 2009

Year 2010

Year 2011

Year 2012

Year 2013

Year 2014

Forecast

Figure 6: Historical power demand and forecast

6.1.2 Robust against theN− k contingency
Typical unit commitment robustness against unit forced outage is achieved by analysing the impact of a

contingency resulting in loss of k generating units. If the total lost generation is greater than the upward
reserve from the remaining generating units, then the system is at risk and demand cannot be fully

sustained. Running such a simulation is an important task and provides insights about the security of the

power system. Unfortunately, if it can prove the inability of the schedule to survive k, the model will fail

to suggest how to improve the schedule to satisfy the safety constraint.

A robust optimization formulation can be used to simulate the loss of k running units and to suggest the
startup and shutdown of new units to safely supply the power demand. The cardinality uncertainty sets

will be used to model the outage state of a unit.

6.2 Mathematical formulation
The setHorizon of consecutive time periods describes the planning horizon of the study. Time periods

t ∈ Horizon have different lengths LENt. Each time period t ∈ Horizon is associated with a power

demandDEMt.

The setUnits denotes the set of available power generation units. Theminimum power generation level

of a unit is PMINu, and its maximum capacity is PMAXu. The startup cost of a single unit isCSTRu, and

the cost of running it at theminimum power generation level isCMINu. Themarginal power production

cost above PMINu for each hour isCADDu.

Each unit u is associated with three decision variables. The binary variable startut equals 1 if the unit u
is starting at the beginning of time period t. The binary variableworkut equals 1 if the unit u is up and

running during the whole time period t. At last, the variable paddut is set to the power generation level

of the unit u during time period t.

6.2.1 Original Unit Commitment Problem

The objective function, to be minimized, is the expected operation cost. The operation cost is composed

of the startup cost and the generation cost. It can be expressed as follows:

∑
u∈Units,t∈Horizon

CSTRu · startut + LENt · (cminuworkut + caddu · paddut)

The startup constraints describe the time period(s) during which the unit is starting. They can expressed

as follows:

startut > workut −workut−1, ∀u ∈ Units, if t > 1

startu1 > worku1 −workuT , ∀u ∈ Units, if t = 0

Themaximum power generation level constraints limit the power generation level of a unit. These

constraints are stated as:

paddut 6 (PMAXu − PMINu) ·workut ∀u ∈ Units, t ∈ Horizon

The power balance constraints ensure that the total power production equals the power demand at any

time. They are formulated by these equations:∑
u∈Units

PMINu ·workut + paddut = DEMu, ∀t ∈ Horizon

6.2.2 Load Robust Unit Commitment Problem

The Load Robust Unit Commitment problem extends the Original Unit Commitment problem by

constraining the unit commitment to be safe under power demand variation. Historical power demands

for the last years are known.

The scenario uncertainty set

LetUdem be the set of possible future power demands. Let Years be the set of past years that should
be taken into account, andHDEMyt the demand for the time period t of the year y. Then the
uncertainty set can be expressed as follows:

Udem = {e : ∀t ∈ Horizon,∃y ∈ Years : et = HDEMyt}

Robust constraints ∑
u∈Units

PMAXu ·workut > demt ∀dem ∈ Udem, t ∈ Horizon

The reformulation engine will efficiently handle the resulting, potentially large number of constraints,

even with large sets of historical data.

6.2.3 The N-k Contingency-Constrained Unit Commitment Problem

TheN− k Contingency-Constrained Unit Commitment problem extends the Original Unit Commitment

problem as to make sure that the committed power generation units can supply the load, even if k
generators are simultaneously forced in outage. The uncertain eut represents the forced outage event,

and the uncertainty set is the set of units that are in forced outage.

The power generation units cannot efficiently be turned on or off at short notice. At the time of

scheduling the units, the demand is not known for sure and wemust make sure that the total maximum

power output capacity of the committed units is greater than the highest demand forecast. Similarly,

the total minimum power generation level must be lower than the lowest demand forecast.

The cardinality uncertainty set

LetUoutage be the set describing groups of combinations of k units in forced outage. This uncertainty
set can be expressed as follows:

Uoutage = {e :
∑

u∈Units

eu 6 k}

Robust constraints∑
u∈Units

PMAXu ∗workut · (1− eu) > DEMt ∀e ∈ Uoutage, t ∈ Horizon

∑
u∈Units

PMINu ∗workut · (1− eu) 6 DEMt ∀e ∈ Uoutage, t ∈ Horizon

These two robust constraints contain the same uncertains. Because uncertains are used to model worst

case situation for a particular constraint, the same uncertain may have different values in each

constraint.

6.3 Implementation

6.3.1 TheOriginal Unit Commitment Implementation

TheMosel code printed below implements and solves the original formulation of the unit commitment

problem.

declarations
NT = 7 ! Number of time periods
TIME = 1..NT ! Time periods
PLANTS = 1..4 ! Power generation plant
UNITS: set of string ! Power generation units

LEN, DEM: array(TIME) of integer ! Length and demand of time periods
PMIN,PMAX: array(PLANTS) of integer ! Min. and max output of a generator type
CSTART: array(PLANTS) of integer ! Start-up cost of a generator
CMIN: array(PLANTS) of integer ! Hourly cost of gen. at min. output
CADD: array(PLANTS) of real ! Cost/hour/MW of prod. above min. level
PLANT: array(UNITS) of integer ! Unit generation plant

YEARS: range ! Historical years
HDEM: array(YEARS,TIME) of integer ! Historical demand profiles

start: array(UNITS,TIME) of mpvar ! Is generation unit starting ?
work: array(UNITS,TIME) of mpvar ! Is generation unit up ?
padd: array(UNITS,TIME) of mpvar ! Production above min. output level
end-declarations

initializations from 'a6electr_ro_simple.dat'
LEN DEM PMIN PMAX CSTART CMIN CADD PLANT HDEM
end-initializations

! Create decision variables
forall(u in UNITS, t in TIME) do
start(u,t) is_binary
work(u,t) is_binary
end-do

! Objective function: total daily cost
Cost:= sum(u in UNITS, t in TIME) (CSTART(PLANT(u))*start(u,t) +
LEN(t)*(CMIN(PLANT(u))*work(u,t) + CADD(PLANT(u))*padd(u,t)))

! Number of generators started per period and per type

forall(u in UNITS, t in TIME)
start(u,t) >= work(u,t) - if(t>1, work(u,t-1), work(u,NT))

! Limit on power production range
forall(u in UNITS, t in TIME)
padd(u,t) <= (PMAX(PLANT(u))-PMIN(PLANT(u)))*work(u,t)

! Power balance
forall(t in TIME)
sum(u in UNITS) (PMIN(PLANT(u))*work(u,t) + padd(u,t)) = DEM(t)

! Symmetry breaker
forall(t in TIME, p in PLANTS)
forall(u1, u2 in UNITS | PLANT(u1) = PLANT(u2) and u1<u2)
work(u1,t) >= work(u2,t)

!**** Solving and solution reporting ****
! Solve the original problem
minimize(Cost)

writeln("\n=== Nominal case ===\n")
print_solution

! Add robust constraints
robust_demand

! Solve robust the problem
setparam("ROBUST_UNCERTAIN_OVERLAP", true)
minimize(Cost)

writeln("\n=== Robust against demand variation ===\n")
print_solution

The following output printing routines are invoked from themodel shown above:

!**** Print highest loss of load in case of worst case scenario realization
procedure print_risk

write("\n", strfmt("Loss of load",20))
forall(t in TIME) do
s:= sum(u in UNITS) (work(u,t).sol*PMAX(PLANT(u))) ! Total capacity
v:= maxlist(DEM(t), max(y in YEARS) HDEM(y,t)) ! Hist. peak demand
if(s < v) then

write(strfmt(v-s,8))
else

write(strfmt("",8))
end-if

end-do
end-procedure

!**** Solution printing ****
procedure print_solution

writeln("Total cost: ", getobjval)

write(strfmt("Time period ",-20))
ct:=0
forall(t in TIME) do
write(strfmt(ct,5), "-", strfmt(ct+LEN(t),2), "")
ct+=LEN(t)

end-do

write("\n",strfmt("Up Units",-20))
forall(t in TIME) write(strfmt(sum(u in UNITS) work(u,t).sol,8))
write("\n", strfmt("Gen. Pwr.",20))
forall(t in TIME)
write(strfmt(sum(u in UNITS) (work(u,t).sol*PMIN(PLANT(u))+padd(u,t).sol),8))

write("\n", strfmt("Gen. Cap.",20))

forall(t in TIME)
write(strfmt(sum(u in UNITS) (work(u,t).sol*PMAX(PLANT(u))),8))

write("\n", strfmt("Res. Dn",20))
forall(t in TIME) write(strfmt(sum(u in UNITS) (padd(u,t).sol),8))
write("\n", strfmt("Res. Up",20))
forall(t in TIME)
write(strfmt(sum(u in UNITS) work(u,t).sol*PMAX(PLANT(u)) - DEM(t),8))

print_risk
writeln

end-procedure

6.3.2 The Load Robust Unit Commitment Implementation

The extension to scenario-based robust optimization is implemented by the following subroutines.

!**** Helper routine to create scenario uncertain set ****
public procedure makescenario(a:array(r1:range,c1:range) of integer,
u:array(c2:range) of uncertain)
declarations
aa:dynamic array(r0:range,U:set of uncertain) of real

end-declarations
forall(i in r1, j in c1|exists(a(i,j)) and exists(u(j)),n as counter) do
aa(n,u(j)):=a(i,j)

end-do
scenario(aa)

end-procedure

!**** Robust against past scenario ****
public procedure robust_demand

declarations
demand: array(TIME) of uncertain ! Uncertain power demand

end-declarations

! Add uncertainty set (time correlation)
makescenario(HDEM,demand)

! Security reserve upward
forall(t in TIME) sum(u in UNITS) PMAX(PLANT(u))*work(u,t) >= demand(t)

end-procedure

6.3.3 TheN-kContingency-ConstrainedUnit Commitment Implemen-

tation

Replacing the call to the procedure robust_demand by a call to the procedure robust_contingency
printed below will turn the problem formulation into a N-k contingency-constrained unit commitment

model.

The example also demonstrates how to use the value of the uncertain to extract meaningful information

about the units which are the most critical ones for the system.

writeln("\n")

!***************************Subroutines***************************
!**** Ensure enough power production capacity allows to
!**** satisfy load even if 'k' units are forced in outage.
procedure robust_contingency(k: integer)

forall(t in TIME, u in UNITS) create(outage(u,t))
forall(t in TIME, u in UNITS) outage(u,t) >= 0
! If the unit is forced in outage, the total capacity

! of the generating units are lost
forall(t in TIME, u in UNITS) outage(u,t) <= 1

! forall(t in TIME) sum(u in UNITS) outage(u,t) <= k
forall(t in TIME) cardinality(union(u in UNITS | exists(outage(u,t))) {outage(u,t)}, k)

forall(t in TIME)
RobCtrMaxDem(t) :=

sum(u in UNITS) PMAX(PLANT(u))*work(u,t) -
sum(u in UNITS) PMAX(PLANT(u))*work(u,t)*outage(u,t) >= DEM(t)

! In order to reduce the number of variables we reuse
! outage() and let the solver do the variable duplication
! by setting the ROBUST_UNCERTAIN_OVERLAP parameter
setparam("ROBUST_UNCERTAIN_OVERLAP",true)
forall(t in TIME)
RobCtrMinDem(t) :=

sum(u in UNITS) PMIN(PLANT(u))*work(u,t) -

6.4 Results
Let us now take a look at the results generated by the three models. The Original Unit Commitment

Problem is used as the baseline scenario and the results of the two robust versions are compared

against it.

For all three model formulations, a feasible unit commitment which satisfies the technical constraints

and supplies the load is found.

The characteristics of the power generation units used in the test instances (number of units per type,

minimum andmaximum generation levels, fixed cost when running at minimum level, variable cost

betweenminimum andmaximum generation levels, start-up cost) are summarized in Table 14.

Table 14: Description of power generators

Unit type Number Pmin Pmax Fix cost Add. MWcost Start-up cost

1 10 750 1750 2250 2.7 5000

2 4 1000 1500 1800 2.2 1600

3 8 1200 2000 3750 1.8 2400

4 3 1800 3500 4800 3.8 1200

Table 15 presents results of the nominal optimization problem. For each time period, the number of

started units is presented (Up Units), along with the generation power (Gen. Pwr.). The generation

capacity (Gen. Cap.) is the total maximum power generation available from the started units. The

columns ’reserve down’ (Res. Dn.) and ’reserve up’ (Res. Up) respectively show the maximum load

decrease or load increase that can be safely supported without requiring startup or shutdown of any

units.

For example, during the first period (0-6) the load can increase from 12000 kWh (power demand

forecast) to 13250 kWh (power demand forecast + reserve up) without having to start up new units.

The last two lines present the maximum loss-of-load in case the worst scenario realizes. For the

demand variation, the worst scenario is the realization of the load profile that requires the highest

power generation increase. For the contingency planning, the worst scenario occurs when the k ’critical’
units are forced in outage. The criticality of a unit depends on its power generation level and total

reserve. A unit without reserve up is critical, as is a unit supplying a very large share of the total load.

Table 15: Nominal case (Total cost: 1.406Million)

0-6 6-9 9-12 12-14 14-18 18-22 22-24

Up units 8 18 15 20 15 17 11

Gen. Pwr. 12000 32000 25000 36000 25000 30000 18000

Gen. Cap. 13250 37750 27250 44750 27250 34250 19250

Res. Dn 3850 10050 8450 10450 8450 9850 6250

Res. Up 1250 5750 2250 8750 2250 4250 1250

Loss of load

Demand variation 0 1395 1491 0 2592 0 0

2 Contingencies 2750 1250 1750 0 1750 2750 2750

Table 16: Robust against demand variation (Total cost: 1.409Million)

0-6 6-9 9-12 12-14 14-18 18-22 22-24

Up Units 8 18 15 20 15 17 11

Gen. Pwr. 12000 32000 25000 36000 25000 30000 18000

Gen. Cap. 13250 39250 28750 44750 30250 37250 19250

Res. Dn 3850 9450 7850 10450 7250 8650 6250

Res. Up 1250 7250 3750 8750 5250 7250 1250

Loss of load 0 0 0 0 0 0 0

If the nominal case is applied, then in case of demand variation, there is a risk of not supplying 1395 kWh

of demand during the morning peak. Common sense would be to start up a new unit in order to cover

the risk for all time periods when there is a risk of loss of load. However, the results from the robust

formulation for the demand variation problem presented in Table 16 show that it is possible to overcome

the risk merely by changing the type of committed units.

Table 17: Robust against 2 contingencies (Total cost: 1.420Million)

0-6 6-9 9-12 12-14 14-18 18-22 22-24

Up Units 9 18 16 20 15 17 11

Gen. Pwr. 12000 32000 25000 36000 25000 30000 18000

Gen. Cap. 19750 39250 32250 44750 33250 38750 25250

Res. Dn 850 9450 6050 10450 6050 8050 3850

Res. Up 7750 7250 7250 8750 8250 8750 7250

Loss of load 0 0 0 0 0 0 0

The loss of load for the 2 contingencies case highlights that there is a risk of not being able to supply

2750 kWh during the initial periods. It means that about 20% of the total demand will not be covered.

Indeed there is only 1250 kWh of upward reserve available, which means that even if just a single unit

running at full capacity is lost (1500 kWh) then the power demand cannot be supplied. Like in the

previous case, the common sense conclusion would be to start up new units. However, the results from

the robust formulation for the 2 contingencies problem presented in Table 17 show that it is possible to

find a unit commitment schedule without risking load disconnection and without committing toomany

new units (and hence incurring higher startup costs).

6.5 References
The presented nominal unit commitment model is based on a problem described in the book

’Applications of optimization with Xpress-MP’ [?].

7 Productionplanningunderenergysupplyuncertainty

7.1 Problem description
A company produces liquid nitrogen and liquid oxygen (we will call them LIN and LOX in the remainder)

and needs to plan production for the nextN periods, identified as shifts of 8 hours each. Although the

resource procurement is rather trivial (these two elements compose most of the troposphere), the

process of obtaining these two liquid gases requires a vast amount of electricity, both for refrigerating

the stored liquid gases and for powering the plant.

Given that the energy cost is the largest component of the production cost, power suppliers offer

several types of energy supply contracts; one of these is known as Interruptible Load Contracts (or ILC

for short) and allows the power supplier to interrupt, with a short notice (a fewminutes), the provision of

electricity to a large customer such as a plant for a limited period of time; most likely the power supplier

will take advantage of this in times of high electricity demand, such as hot summer days. However, the

ILC requires that there be nomore than K interruptions throughout the planning horizon.

We are dealing therefore with a production planning problem under uncertainty in the power supply.

We are given in input the production and inventory cost, the maximum production, the inventory

capacity, the initial inventory, the demand in LIN and LOX for each period, and the maximum number K
of interruptions. The problem consists of finding the amount of LIN and LOX to be produced every day

so that the demands are satisfied irrespective of the interruptions that may occur within the clause of

the contract.

This optimization problem requires the application of Robust Optimization for one reason among

several: among the customers of this company there are hospitals, for which the fulfilment of the

demand of LOX is imperative.

7.2 Mathematical formulation
We are given the set of gasesG = {lin, lox} and the set of time periods T = {1, 2, ...,N}, the maximum

production level Plin and Plox for each period and the inventory capacity Slin and Slox for each gas.
The demand is given by the parameter demtg for t ∈ T ,g ∈ G.

The uncertainty set is defined by every vector of interruptions (ξ)i∈T such that ξi ∈ [0, 1] and not more

than K of its elements are positive:

∑
i∈T

ξi 6 K.

Let us define the variables prodtg and invtg representing the production and inventory level,
respectively, at period t for gas g. Without uncertainty in the input data, themodel would have simple of

constraints: bounds on the variables and fixing of the initial inventory level:

0 6 prodtg 6 Pg ∀g,∀t ∈ {1, 2, ...,N}

0 6 invtg 6 Sg ∀g,∀t ∈ {1, 2, ...,N}

0 6 inv0g = Ig ∀g;

and the production planning constraint would consist of a mass conservation constraint for every gas g
and time period t:

invt−1,g + prodtg = invtg + demtg ∀t ∈ T ,g ∈ G.

The modeling is more involved for this case. A first attempt would be to write the robust constraint by

multiplying the production variable by (1− ξt), so that the actual production at time period t is zero if
ξt = 1:

invt−1,g + (1− ξt)prodtg = invtg + demtg ∀t ∈ T ,g ∈ G.

However, this would yield no feasible solution: for each robust constraint related to production at time

period t the uncertain ξt would be set to one, and no production would occur. We need to aggregate the

production constraints in such a way that the balance between production and demand yields an

inventory that is nonnegative at every time period. The above constraint is replaced by the following

one:

inv0,g +
∑

τ∈T :τ6t

((1− ξτ)prodτg − demτg) > 0 ∀t ∈ T ,g ∈ G.

This robust constraint considers uncertainty in the interruptions but allows for a production plan that

satisfies all demands even with K interruptions. In order to deal with contour conditions and take into

account the initial inventory, we add a slight modification to the uncertainty set and require that no

interruption happens at time period 1:

ξ1 = 0.

Finally, the robust value of the inventory must take into account the situation in which, after planning the

production, no interruption actually occurs: this is simply enforced by bounding the inventory at every

time period as the sum of all accumulated production discounted by the accumulated demand:

invtg > inv0,g +
∑

τ∈T :τ6t

(prodτg − demτg) ∀t ∈ T ,g ∈ G.

7.3 Implementation
Below is the implementation in Mosel: note that the option ROBUST_OVERLAP_UNCERTAIN is again

used since the nonnegative-inventory constraints use the same set of uncertains multiple times.

model robust_prodplan
uses "mmrobust"

parameters
plan_data = "prodplan_robust.dat"

end-parameters

declarations
NDAYS: integer ! Planning horizon
PERDAY: integer ! Number of periods per day
NPERIODS: integer ! Total number of periods

PERIODS,PERIODS0: range ! Time periods
GASES: set of string ! Set of products

PROD_CAP: array(GASES) of real ! Production capacity every day
INV_CAP: array(GASES) of real ! Inventory capacity
INV_0: array(GASES) of real ! Initial inventory

PROD_COST: real ! Production cost
INV_COST: real ! Inventory cost

DEMAND: array (PERIODS, GASES) of real ! Demand of each gas every day

MAX_NINTERR: integer ! Maximum number of interruptions
! (as per contract)

end-declarations

!**** Initialize data ****
initializations from plan_data
NDAYS PERDAY
PROD_CAP INV_CAP
PROD_COST INV_COST INV_0
DEMAND
MAX_NINTERR

end-initializations

NPERIODS := NDAYS * PERDAY
PERIODS0 := 0..NPERIODS

!**** Problem formulation ****
declarations
produce: array(PERIODS, GASES) of mpvar ! Production every day
inventory: array(PERIODS0, GASES) of mpvar ! Inventory level every day,

! including initial level

interruption: array(PERIODS) of uncertain ! Is power cut at this time?
RobProd: array(PERIODS, GASES) of robustctr ! Robust constraint dealing with uncertainty

end-declarations

!**** Constraints ****

! Start inventory
forall(g in GASES)
inventory (0,g) = INV_0 (g)

! Inventory balance
forall(t in PERIODS, g in GASES) do

RobProd(t,g) := inventory(0,g) + sum(tp in PERIODS | tp <= t)
((1 - interruption(tp)) * produce(tp,g) - DEMAND(tp,g)) >= 0

inventory (0,g) + sum (tp in PERIODS | tp <= t)
(produce(tp,g) - DEMAND(tp,g)) <= inventory(t,g)

inventory(t,g) <= INV_CAP(g)
produce(t,g) <= PROD_CAP(g)

end-do

! Interruptions of production
forall (t in PERIODS) do
interruption (t) <= 1
interruption (t) >= 0

end-do

sum(t in PERIODS) interruption (t) <= MAX_NINTERR
interruption(1) = 0

!**** Solving ****
setparam("robust_uncertain_overlap", true)

! Set verbosity level
setparam("xprs_verbose", true)

! Objective function: total cost of production and storage

minimize(sum (t in PERIODS, g in GASES)
(PROD_COST * produce (t,g) + INV_COST * inventory(t,g)))

!**** Solution reporting ****
writeln("\nNumber of interruptions: ", MAX_NINTERR)
writeln("\nOptimal solution has cost ", getobjval)

COLWIDTH := 6

forall(g in GASES) do

writeln("\nProduction of ", g)

write(strfmt ("Time",-COLWIDTH))
forall(t in PERIODS0) write (strfmt(t,COLWIDTH))

write("\n", strfmt("Dem",-2*COLWIDTH))
forall(t in PERIODS) write(strfmt(DEMAND(t,g),COLWIDTH,1))

write("\n", strfmt("Prod",-2*COLWIDTH))
forall(t in PERIODS) write(strfmt (produce(t,g).sol,COLWIDTH,1))

write("\n", strfmt("Inv*",-COLWIDTH))
forall(t in PERIODS0)
write (strfmt(inventory(0,g).sol +

sum (tp in PERIODS | tp <= t)
(produce(tp,g).sol - DEMAND(tp,g)),COLWIDTH,1))

writeln("\n\nWorst-case interruptions for ", g, ": ")

forall(t in PERIODS | getsol(interruption(t),RobProd(t,g)) > 0)
write (t, " ")

writeln
end-do

end-model

7.4 Input Data
For this example, we consider a simple instance of the problemwhere the planning has a horizon of five

days and three periods per day. The cost of production and inventory is 4 and 3, respectively, and the

ILC contract allows for four interruptions. Table 18 below shows the production and inventory capacity

and the initial inventory level for both gases.

Table 18: Production and inventory capacity

Gas Prod. capacity Inv. capacity Initial inv.

LIN 29 60 20

LOX 5.5 27 20

Table 19 below lists the demand for each gas over all 15 periods.

Table 19: Customer demand

Gas/Time 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15

LIN 6 14 10 8 11 15 10 9 10 9 10 12 11 15 9

LOX 2 5 3 4 8 4 8 7 5 4 3 3 5 9 7

7.5 Results
The optimal solution of this problem has cost 4727.17. Tables 20 and 21 give the optimal levels of

production required in order to satisfy the demands regardless of when the interruptions occur, if ever.

Note that the production levels are higher at the beginning and then, for liquid oxygen, they equal the

demand. This depends on the fact that production at the initial time periods must assume for

worst-case scenarios such as K = 4 consecutive interruptions early in the planning horizon. The

inventory levels equal the balance between demand and production as they must account for the event

that no interruption occurs.

Table 20: Production of LIN

Time 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15

Dem 6 14 10 8 11 15 10 9 10 9 10 12 11 15 9

Prod 29 15 15 15 15 15 10 9 10 9 10 12 11 15 9

Inv* 43 44 49 56 60 60 60 60 60 60 60 60 60 60 60

Table 21: Production of LOX

Time 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15

Dem 2 5 3 4 8 4 8 7 5 4 3 3 5 9 7

Prod 5.5 5.2 5.2 5.2 5.2 5.2 5.2 5.2 5 5.2 5.2 5.2 5.2 5.2 5.2

Inv* 23.5 23.7 25.8 27 24.2 25.3 22.5 20.7 20.7 21.8 24 26.2 26.3 22.5 20.7

Note that this is the result of the optimization problem solved at the beginning of the time period. Using

a rolling horizon approach, i.e., re-solving the model at every time period while taking into account the

interruptions that already occurred, would allow us to obtain a less expensive production plan although

the planned production levels might have to change.

Finally, since the interruption of a plant at a given period affects the production of both gases, the

retrieval of uncertains needs to be done for each gas and getsol takes both interruption and the
robust constraint RobProd as arguments. As would be expected, most worst-case interruptions are at

the initial time periods, since those are critical moments when the LIN and LOX tanks need to be built up.

Worst-case interruptions for LIN:
2 3 4 5 6 14

Worst-case interruptions for LOX:
2 3 4 5 6 8 10 15

7.6 References
This example is a simplified version of a real-world application that has been documented in [?].

